

FeynArts 3.5 User's Guide

October 19, 2010 Thomas Hahn

The dreadful legal stuff: *FeynArts* is free software, but is not in the public domain. Instead it is covered by the GNU library general public license. In plain English this means:

- 1) We don't promise that this software works. (But if you find any bugs, please let us know!)
- 2) You can use this software for whatever you want. You don't have to pay us.
- 3) You may not pretend that you wrote this software. If you use it in a program, you must acknowledge somewhere in your documentation that you've used our code.

If you're a lawyer, you will rejoice at the exact wording of the license at <http://www.fsf.org/copyleft/lgpl.html>.

FeynArts is available from <http://www.feynarts.de>. If you make this software available to others, please provide them with this manual, too. There exists a low-traffic mailing list where updates will be announced. Contact hahn@feynarts.de to be added to this list.

If you find any bugs, or want to make suggestions, or just write fan mail, address it to:

Thomas Hahn
Max-Planck-Institut für Physik
(Werner-Heisenberg-Institut)
Föhringer Ring 6
D-80805 Munich, Germany
e-mail: hahn@feynarts.de

Contents

1	Getting Started	5
2	Roadmap of <i>FeynArts</i>	8
3	Creating the Topologies	9
3.1	Topological Objects	9
3.2	CreateTopologies	10
3.3	Creating Counter-term Topologies	11
3.4	Creating Topologies with generic Vertex Functions	12
3.5	Excluding Topologies	13
3.6	Selecting Starting Topologies	17
4	Inserting Fields into Topologies	20
4.1	The Three-Level Fields Concept	20
4.2	InsertFields	22
4.3	Model Files	25
4.4	Imposing Restrictions	26
4.5	Selecting Insertions	28
4.6	Grouping Insertions	29
4.7	Auxiliary Functions	29
4.8	Discarding Insertions	31
4.9	Modifying Insertions	32
4.10	Structure of the Inserted Topologies	33
5	Drawing Feynman Graphs	34
5.1	Things to do with the Paint Output	36
5.2	Shaping Topologies	37
6	Creating the Analytic Expressions	40
6.1	Representation of Feynman Amplitudes	40

6.2	CreateFeynAmp	42
6.3	Interpreting the Results	45
6.4	Picking Levels	46
6.5	General Structure of the Amplitudes	47
6.6	On Fermion Chains and Signs	48
6.7	Specifying Momenta	49
6.8	Compatibility with <i>FeynArts</i> 1	49
7	Definition of a New Model	50
7.1	The Generic Model	50
7.2	The Classes Model	57
7.3	Add-on Model Files	66
7.4	Model-file Generation	68
A	The Lorentz Formalism	71
B	The Electroweak Standard Model	73
B.1	The QCD Extension	75
B.2	Background-field Formalism	76
C	The Minimal Supersymmetric Standard Model	77
D	The Two-Higgs-Doublet Model	80
E	Graphics Primitives in <code>feynarts.sty</code>	81
E.1	Geometry	81
E.2	Propagators	82
E.3	Vertices	83
E.4	Labels	84
F	Incompatible Changes in Version 3.3	85

1 Getting Started

FeynArts is a *Mathematica* package for the generation and visualization of Feynman diagrams and amplitudes. It started out in 1990 as a Macsyma code written by Hagen Eck and Sepp Küblbeck which could produce tree-level and one-loop diagrams in the Standard Model [Kü90], but soon got ported to the *Mathematica* platform. In 1995, Hagen Eck designed the second version to be a fully general diagram generator. To achieve this, he implemented some decisive new ideas [Eck95], the most important one being the generation of diagrams in three levels. The program was taken up again in 1998 by Thomas Hahn who developed version 2.2. The well-designed conceptual framework was kept, but the actual code was reprogrammed almost entirely to make it more efficient and a user-friendly topology editor was added. The current version 3 features a completely new rendering engine for PostScript and \LaTeX , together with full support of the *Mathematica* Frontend's graphical capabilities. It is also no longer dependent on the X platform for topology editing.

The main features of *FeynArts* are:

- The generation of diagrams is possible at three levels: generic fields, classes of fields, or specific particles.
- The model information is contained in two special files: The *generic model file* defines the representation of the kinematical quantities like spinors or vector fields. The *classes model file* sets up the particle content and specifies the actual couplings. Since the user can create own model files, the applicability of *FeynArts* is virtually unlimited within perturbative quantum field theory. As generic model the Lorentz formalism (`Lorentz.gen`) and as classes model the electroweak Standard Model in several variations (`SM.mod`, `SMQCD.mod`, `SMbgf.mod`), the Minimal Supersymmetric Standard Model (`MSSM.mod`, `MSSMQCD.mod`), and the Two-Higgs-Doublet Model (`THDM.mod`) are supplied.
- In addition to ordinary diagrams, *FeynArts* can generate counter-term diagrams and diagrams with placeholders for one-particle irreducible vertex functions (skeleton diagrams).
- *FeynArts* employs the so-called “flipping-rule” algorithm [De92] to concatenate fermion chains. This algorithm is unique in that it works also for Majorana

fermions and the fermion-number-violating couplings they entail (e.g. quark–squark–gluino) and hence allows supersymmetric models to be implemented.

- Restrictions of the type “field X is not allowed in loops” can be applied. This is necessary e.g. for the background-field formulation of a field theory.
- Vertices of arbitrary adjacency, required for effective theories, are allowed.
- Mixing propagators, such as appear in non- R_ξ -gauges, are supported.
- *FeynArts* produces publication-quality Feynman diagrams in PostScript or \LaTeX in a format that allows easy customization.

These features have been introduced in version 2 but have received some in parts considerable improvements in version 3. The user interface, on the other hand, has through all versions suffered only minor and mostly backward-compatible changes, and the major functions can still be used in essentially the same way as in version 1.

Installation

FeynArts requires *Mathematica* 3.0 or above. In *Mathematica* versions before 5.0, a Java VM and the J/Link package are needed for the topology editor. Both ingredients can be obtained free of charge from

<http://www.wolfram.com/solutions/mathlink/jlink> (J/Link),
<http://java.sun.com/j2se> (Java).

Note that many systems (e.g. Windows) have a Java VM pre-installed. Follow the instructions that come with J/Link for installation on the various platforms.

FeynArts comes in a compressed tar archive *FeynArts-n.m.tar.gz* which merely needs to be unpacked, no further installation is necessary:

```
gunzip -c FeynArts-n.m.tar.gz | tar xvf -
```

Unpacking the archive creates a subdirectory *FeynArts-n.m* which contains

FeynArts.m

the main program


Setup.m	initialization file
FeynArts/	directory containing the <i>FeynArts</i> code
Models/	directory containing the model files
ShapeData/	directory containing the shapes of topologies
README	additional information about this release
HISTORY	general blurb about the evolution of <i>FeynArts</i>
Convert2to3.m	conversion program for <i>FeynArts</i> 2 GraphInfo files
Convert3to31.m	conversion program for <i>FeynArts</i> 3 GraphInfo files
Convert31to32.m	conversion program for <i>FeynArts</i> 3.1 GraphInfo files

Permanent changes of parameters, options, etc. should be placed in Setup.m. Patching the *FeynArts* code directly is not recommended since it is inherently unportable.

Several *FeynArts* functions have options that take a list of objects. Except in the case of level specifications (see Sect. 4.1), the list may be omitted if it contains only one element, e.g. `ExcludeTopologies -> Tadpoles` instead of `ExcludeTopologies -> {Tadpoles}`.

Some *FeynArts* functions write messages to the screen to indicate their progress. These messages can be partially or completely suppressed by setting `$Verbose` to 0 (no messages) or 1 (summary messages only). The default is 2 (all messages).

2 Roadmap of *FeynArts*



3 Creating the Topologies

3.1 Topological Objects

For the purposes of *FeynArts*, a *topology* is a set of lines (propagators) connecting a set of points (vertices). Furthermore, topologies in *FeynArts* are restricted to be *connected* topologies, where every part of the topology is connected to the rest with at least one propagator.

A vertex is characterized by two numbers: its adjacency and its counter-term order. The *adjacency* is the number of propagators that run into the vertex.

Vertex[<i>adj</i>] [<i>n</i>]	vertex with adjacency <i>adj</i> , counter-term order 0, and number <i>n</i>
Vertex[<i>adj</i> , <i>cto</i>] [<i>n</i>]	vertex with adjacency <i>adj</i> , counter-term order <i>cto</i> , and number <i>n</i>

A propagator connects two vertices, possibly carrying a field.

Propagator[<i>t</i>] [<i>from</i> , <i>to</i>]	propagator of type <i>t</i> running from <i>from</i> to <i>to</i>
Propagator[<i>t</i>] [<i>from</i> , <i>to</i> , <i>field</i>]	propagator of type <i>t</i> running from <i>from</i> to <i>to</i> carrying field <i>field</i>
<i>possible types of propagators:</i>	
Incoming, Outgoing	external propagator flowing in or out
External	undirected external propagator
Internal	internal propagator which is not part of a loop
Loop[<i>n</i>]	internal propagator on loop <i>n</i> *

The propagators, then, are collected into topologies.

Topology[<i>p</i> ₁ , <i>p</i> ₂ , ...]	representation of a topology with propagators <i>p</i> _{<i>i</i>}
Topology[<i>s</i>] [...]	the same with combinatorial factor 1/ <i>s</i>
TopologyList[<i>t</i> ₁ , <i>t</i> ₂ , ...]	a list of topologies <i>t</i> _{<i>i</i>}
TopologyList[<i>info</i>] [...]	the same with an additional information field

*The *n* in Loop[*n*] is not the actual number of the loop—which in general cannot be determined unambiguously—but the number of the one-particle irreducible conglomerate of loops.

3.2 CreateTopologies

The basic function to generate topologies is `CreateTopologies`. It creates topologies based on how many loops and external legs they have.

<code>CreateTopologies[l, i -> o]</code>	create topologies with <i>l</i> loops, <i>i</i> incoming, and <i>o</i> outgoing legs
<code>CreateTopologies[l, e]</code>	create topologies with <i>l</i> loops and <i>e</i> external legs

Load *FeynArts*.


```
In[1]:= << FeynArts`
FeynArts 3.3
by Hagen Eck, Sepp Kueblbeck, and Thomas Hahn
last revised 12 Oct 06
```

Create all topologies with one loop, one incoming, and one outgoing line. The results are collected in a `TopologyList`.

```
In[2]:= CreateTopologies[1, 1 -> 1]
Out[2]:= TopologyList[
  Topology[2] [
    Propagator[Incoming] [Vertex[1] [1], Vertex[4] [3]],
    Propagator[Outgoing] [Vertex[1] [2], Vertex[4] [3]],
    Propagator[Loop[1]] [Vertex[4] [3], Vertex[4] [3]] ]
  Topology[2] [
    Propagator[Incoming] [Vertex[1] [1], Vertex[3] [3]],
    Propagator[Outgoing] [Vertex[1] [2], Vertex[3] [3]],
    Propagator[Internal] [Vertex[3] [3], Vertex[3] [4]],
    Propagator[Loop[1]] [Vertex[3] [4], Vertex[3] [4]] ],
  Topology[2] [
    Propagator[Incoming] [Vertex[1] [1], Vertex[3] [3]],
    Propagator[Outgoing] [Vertex[1] [2], Vertex[3] [4]],
    Propagator[Loop[1]] [Vertex[3] [3], Vertex[3] [4]],
    Propagator[Loop[1]] [Vertex[3] [3], Vertex[3] [4]] ] ] ]
```

The painted version of these topologies is much easier to understand than the list form.

```
In[3]:= Paint[%]
```



CreateTopologies uses a recursive algorithm that generates topologies with n legs from topologies with $n - 1$ legs [Kü90]. The recursion iterates down to zero external legs where it is stopped by a pre-defined set of *starting topologies*.

Several options influence the behaviour of CreateTopologies:

<i>option</i>	<i>default value</i>	
Adjacencies	{3, 4}	allowed adjacencies of the vertices
CTOrder	0	counter-term order of the topologies
ExcludeTopologies	{}	list of filters for excluding topologies
StartingTopologies	All	list to starting topologies to use

Adjacencies gives the allowed adjacencies the vertices may have. For example, Adjacencies -> 4 generates only topologies with 4-vertices. In renormalizable quantum field theories, 3 and 4 are the only possible adjacencies.

CTOrder specifies at which counter-term order topologies are generated. Note that CreateTopologies creates counter-term topologies for *exactly* the given counter-term order.

3.3 Creating Counter-term Topologies


Beyond one loop, one generally needs counter-terms of more than one order, e.g. in a two-loop calculation the second-order counter-terms on tree topologies as well as the first-order counter-terms on one-loop topologies are needed. This more comprehensive task is handled by CreateCTTopologies. Its options are the same as for CreateTopologies except that CTOrder is ignored.

CreateCTTopologies [$l, i \rightarrow o$]	create all counter-term topologies up to order l with i incoming and o outgoing legs
CreateCTTopologies [l, e]	create all counter-term topologies up to order l with e external legs

Once again: use CreateTopologies to generate topologies with l loops and counter-term order cto ; use CreateCTTopologies to generate counter-term topologies for calculations of order l . Specifically, CreateCTTopologies [$l, i \rightarrow o$] creates all counter-term topologies needed for the topologies created by CreateTopologies [$l, i \rightarrow o$].

3.4 Creating Topologies with generic Vertex Functions

Topologies can also be created with “placeholders” for one-particle irreducible (1PI) vertex functions. Such topologies are sometimes also called skeleton diagrams. The vertex-function placeholders are represented graphically by a grey bubble, e.g.



The idea is to reduce the number of diagrams by calculating the vertex functions separately and inserting the final expression into tree diagrams at the proper places. For example, in a $2 \rightarrow 2$ process, the self-energy diagrams are generated once for the s -channel, once for the t -channel, and once for the u -channel. It would of course be much more economic to calculate the necessary 1PI two-point vertex functions only once, and insert them at the proper places in the s -, t -, and u -channel tree diagrams. The problem may not be particularly acute in this example at one loop, but it can easily become significant for higher loop order or more external legs.

Topologies with vertex-function placeholders are generated by `CreateVFTopologies`. Note that there is always a loop order associated with the vertex function, hence one has to specify a loop order also for these “VF” topologies.

<code>CreateVFTopologies[l, i->o]</code>	create topologies with placeholders for 1PI vertex functions of order l with i incoming and o outgoing legs
<code>CreateVFTopologies[l, e]</code>	create topologies containing generic 1PI vertex-function insertions of order l with e external legs

The creation of “VF” topologies and of counter-term topologies is essentially the same. In fact, the place where the vertex function must later be inserted is given precisely by the location of the cross on the counter-term diagram of the corresponding order.

`CreateFeynAmp` translates the vertex-function placeholders into generic objects of the form `VertexFunction[o][f1, f2, ...]` which represent the 1PI vertex function $\Gamma_{f_1 f_2 \dots}^{(o)}$, where o is the loop order and f_1, f_2, \dots are the adjoining fields (direction: all incoming), complete with their momenta and kinematic indices. The conventions are such that a `VertexFunction` consists directly of the corresponding 1PI diagrams (as generated by *FeynArts*) without further prefactors.

Internally, the tricky part is to decide which vertex functions are allowed and which are not. Unlike in the case of ordinary counter-terms, `InsertFields` cannot simply look up which vertices are present in the model since there are exceptions: for instance in the electroweak Standard Model the $\gamma\gamma H$ vertex has neither a counter-term nor even a tree-level vertex. Nevertheless, loop diagrams for this vertex exist.

Therefore, different constraints have to be used: only such vertex functions are generated for which there exists a corresponding generic vertex, and which do not violate conservation of quantum numbers. The quantum numbers of the fields are defined in the classes model file (see Sect. 7.2).

3.5 Excluding Topologies


The `ExcludeTopologies` option specifies topology exclusion filters. Such a filter is a special function whose outcome when applied to a topology—True or not—determines whether the topology is kept or discarded. Some filters are supplied with *FeynArts*, others can be defined. The pre-defined filters work on topologies of any loop number.

		<i>exclude topology if it contains...</i>
Loops [<i>patt</i>]	loops	of adjacency <i>patt</i>
CTs [<i>patt</i>]	counter-terms	
Tadpoles	loops	of adjacency 1
TadpoleCTs	counter-terms	
SelfEnergies	loops	of adjacency 2
SelfEnergyCTs	counter-terms	
Triangles	loops	of adjacency 3
TriangleCTs	counter-terms	
Boxes	loops	of adjacency 4
BoxCTs	counter-terms	
Pentagons	loops	of adjacency 5
PentagonCTs	counter-terms	
Hexagons	loops	of adjacency 6
HexagonCTs	counter-terms	
AllBoxes	loops	of adjacency ≥ 4
AllBoxCTs	counter-terms	
WFCorrections	self-energy or tadpole	loops
WFCorrectionCTs		counter-terms on ext. legs
WFCorrections [<i>patt</i>]	ditto only on ext. legs matching <i>patt</i>	
WFCorrectionCTs [<i>patt</i>]		
Internal	propagators of type Internal, i.e. if the topology is one-particle reducible	

Because the generation of counter-term topologies and topologies with vertex-function insertions (see Sect. 3.4) is so similar, there are no special exclusion filters for the latter—one simply uses the filters for counter-terms.


Create all $1 \rightarrow 2$
topologies.

```
In[4]:= CreateTopologies[1, 1 -> 2];
In[5]:= Paint[%, ColumnsXRows -> 4]
```



Now the same for
irreducible topologies.

```
In[6]:= CreateTopologies[1, 1 -> 2,
ExcludeTopologies -> Internal];
In[7]:= Paint[%, ColumnsXRows -> 4]
```



For convenience, common choices of the `ExcludeTopologies` option have a short-cut:

	<i>same as...</i>
<code>TadpolesOnly</code>	<code>ExcludeTopologies -> Loops[Except[1]]</code>
<code>TadpoleCTsOnly</code>	<code>ExcludeTopologies -> CTs[Except[1]]</code>
<code>SelfEnergiesOnly</code>	<code>ExcludeTopologies -> {Loops[Except[2]], WFCorrections}</code>
<code>SelfEnergyCTsOnly</code>	<code>ExcludeTopologies -> {CTs[Except[2]], WFCorrectionCTs}</code>
<code>TrianglesOnly</code>	<code>ExcludeTopologies -> Loops[Except[3]]</code>
<code>TriangleCTsOnly</code>	<code>ExcludeTopologies -> CTs[Except[3]]</code>
<code>BoxesOnly</code>	<code>ExcludeTopologies -> Loops[Except[4]]</code>
<code>BoxCTsOnly</code>	<code>ExcludeTopologies -> CTs[Except[4]]</code>
<code>PentagonsOnly</code>	<code>ExcludeTopologies -> Loops[Except[5]]</code>
<code>PentagonCTsOnly</code>	<code>ExcludeTopologies -> CTs[Except[5]]</code>
<code>HexagonsOnly</code>	<code>ExcludeTopologies -> Loops[Except[6]]</code>
<code>HexagonCTsOnly</code>	<code>ExcludeTopologies -> CTs[Except[6]]</code>

To extend the filtering capabilities, you may define your own filter functions. Here is how the `Triangles` filter is defined:

```
$ExcludeTopologies[ Triangles ] = FreeQ[ToTree[#], Centre[3]]&
```

The filter function *must* be defined as a pure function (`foo[#]&`) since it will be grouped together with other filter functions by `CreateTopologies`. The function will be passed a topology as its argument. This is important to know for structural operations, e.g. the following filter excludes topologies with 4-vertices on external legs:

```
$ExcludeTopologies[ V4onExt ] =  
  FreeQ[ Cases[#, Propagator[External][_]], Vertex[4] ]&
```


<code>\$ExcludeTopologies[name] = func[#]&</code>	defines the filter <i>name</i>
<code>ToTree[top]</code>	returns the topology <i>top</i> with each loop shrunk to a point named <code>Centre[adj][n]</code> where <i>adj</i> is the adjacency of loop <i>n</i>
<code>Centre[adj][n]</code>	represents the remains of loop <i>n</i> with adjacency <i>adj</i> after it has been shrunk to a point by <code>ToTree</code>


3.6 Selecting Starting Topologies

Using the option `StartingTopologies` to select particular starting topologies may significantly speed up `CreateTopologies`, especially for higher-loop topologies. For example, it is much faster to start without the reducible three-loop starting topologies than to exclude the reducible topologies afterwards. The default setting `All` evaluates to all starting topologies for the given loop number and counter-term order. It is allowed to use patterns, e.g. `StartingTopologies -> Three[_]` selects only the irreducible three-loop starting topologies.

<code>Theta, Eight, Bicycle</code>	two-loop starting topologies
<code>Three[1...8]</code>	irreducible three-loop starting topologies
<code>ThreeRed[1...7]</code>	reducible three-loop starting topologies
<code>CT[l, cto][n]</code>	counter-term starting topologies, currently defined for loop number and counter-term order $(l, cto) = (0, 2), (0, 3), (1, 1), (1, 2), \text{ and } (2, 1)$
<code>StartTop[l, cto]</code>	the starting topologies for loop number <i>l</i> and counter-term order <i>cto</i>
<code>SymmetryFactor[t]</code>	find the combinatorial factor of the starting topology <i>t</i>

You can also draw the starting topologies. The cross marks a first-order and the circled cross a second-order counter-term.

```
In[8]:= Paint[StartTop[1, 2]]
```



Up to three loops all starting topologies including the counter-terms are supplied with *FeynArts*. If you need others, you must enter them yourself: Edit `Topology.m` and locate the definition of `StartTop`. There more starting topologies can be appended. Since entering starting topologies is not an everyday job, some restrictions have been imposed that enable *FeynArts* to work with much faster algorithms.

- 1) There is an important distinction between positive and negative vertex identifiers (the v in `Vertex[e][v]`). Vertices with negative identifiers are so-called permutable vertices. They are used for weeding out topologically equivalent topologies. The algorithm is roughly the following: The topologies are sorted into some canonical order, and then compared. This simple method, however, fails whenever a graph has a symmetry. In that case, the indices of the symmetrical vertices have to be permuted to give all topologically equivalent versions. It is this “power set” of each topology that is actually compared. If you’re not sure which vertices should be permutable, make them *all* permutable. This will be slower, but safer.

- 2) For the correct functioning of the `ExcludeTopologies` filters it is essential that the propagators on an irreducible conglomerate of loops have the *same* loop number (the n in `Loop[n]`), no matter how many loops there actually are. For example, the two-loop starting topology `Theta` has only `Loop[1]` propagators.
- 3) Vertex identifiers must always be unique, e.g. having both a `Vertex[3][1]` and a `Vertex[4][1]` within the same topology is forbidden.
- 4) To give the starting topology a name which can be used with `StartingTopologies`, define it with `define[name] = Topology[s][...]`. To make *name* accessible outside of `Topology.m` it must either be declared in `FeynArts.m` via `name::usage` or else live in the `Global`` context (i.e. `define[Global`name] = ...`). If there is only one starting topology, or one always wants to use all of the starting topologies, the `define[...]` can be omitted.
- 5) To determine the (inverse) symmetry factor (the s in `Topology[s][...]`), enter the topology with an arbitrary factor first (e.g. `Topology[1][...]`), then apply `SymmetryFactor` to find the right symmetry factor, and with it supplement the initial definition.

For example, the two-loop starting topologies are defined as

```
StartTop[2, 0] = TopologyList[
  define[Theta] = Topology[12][
    Propagator[Loop[1]][Vertex[3][-2], Vertex[3][-1]],
    Propagator[Loop[1]][Vertex[3][-2], Vertex[3][-1]],
    Propagator[Loop[1]][Vertex[3][-2], Vertex[3][-1]] ],
  define[Eight] = Topology[8][
    Propagator[Loop[1]][Vertex[4][1], Vertex[4][1]],
    Propagator[Loop[1]][Vertex[4][1], Vertex[4][1]] ],
  define[Bicycle] = Topology[8][
    Propagator[Internal][Vertex[3][-2], Vertex[3][-1]],
    Propagator[Loop[1]][Vertex[3][-2], Vertex[3][-2]],
    Propagator[Loop[2]][Vertex[3][-1], Vertex[3][-1]] ]
]
```

4 Inserting Fields into Topologies

4.1 The Three-Level Fields Concept

FeynArts distinguishes three different levels of fields, generic fields, classes of fields, and specific particles. Field information becomes more and more specific with these levels.

Generic, Classes, Particles	field levels in <i>FeynArts</i>
-----------------------------	---------------------------------

Generic fields are the abstract field types.

F, S, V, U, T	basic field types: fermion, scalar, vector, ghost, and tensor fields
SV	scalar–vector mixing field

Classes fields represent sets of fields with common properties such as behaviour under charge conjugation. A class is a generic field type with a class number, e.g. $F[1]$. The class specifies which further indices (if any) the class members possess and the range of these indices.

Particles fields are then class members with definite indices, e.g. $F[1, \{1, 2\}]$ if the class $F[1]$ has two indices. For classes fields without further indices, classes and particles fields are the same.

Antiparticles (charge-conjugate fields) are denoted by a minus sign in front of the field, e.g. if $F[2, \{1\}]$ is the electron, $-F[2, \{1\}]$ is the positron.

Apart from simple fields, *FeynArts* can also handle mixing fields. A mixing field propagates like any other field, but has no couplings of its own. Instead, it couples like one simple field on the left side and like another simple field on the right side, e.g. if the scalar–vector mixing field $SV[3]$ has the mixing partners $\{S[3], V[3]\}$, it couples as if it were an $S[3]$ on the left and a $V[3]$ on the right.

Whereas simple fields can have at most two states, the field and its antifield, a mixing field can occur in four states, the mixing field, its antifield, the reversed mixing field, and its antifield. Unlike for simple fields, the antifield of $SV[3]$ is $-2SV[3]$ and the antifield of $-SV[3]$ is $2SV[3]$. Self-conjugate fields cannot have negative coefficients, i.e. in that case the two possibilities in the bottom row are absent:

$$\begin{array}{cc}
 \overrightarrow{\text{SV}[3]} & 2 \overrightarrow{\text{SV}[3]} \\
 \overleftarrow{-\text{SV}[3]} & -2 \overleftarrow{\text{SV}[3]}
 \end{array}$$

Scalar–vector mixing is a special case because it requires different handling internally already at generic level. It is disabled by default.

`$SVMixing = True (False)` enable (disable) scalar–vector mixing fields
(must be set *before* the model is initialized)

At generic level there are no antiparticles or reversed mixing propagators. (Exception: VS is used as the reversed SV field.)

F	generic fermion field
F[n]	fermion class <i>n</i> , e.g. F[2] is the class of leptons in SM.mod
F[n, {i, ...}]	member of fermion class <i>n</i> , e.g. F[2, {1}] is the electron in SM.mod
-field	charge-conjugate of <i>field</i>
2 mixingfield	<i>mixingfield</i> with left and right partner reversed

Using different levels of fields is a natural concept in perturbative field theory. The kinematical structure of a coupling is determined at the generic level. Consider the scalar–fermion–fermion coupling in the Lorentz formalism. Its kinematical structure is

$$C_{\text{FFS}} = G_{\text{FFS}}^{\omega_-} \omega_- + G_{\text{FFS}}^{\omega_+} \omega_+$$

(ω_{\pm} being the chirality projectors). $G_{\text{FFS}}^{\omega_-}$ and $G_{\text{FFS}}^{\omega_+}$ are generic coupling constants which carry two kinds of indices: the fields they belong to and the kinematical object they appear with. At classes level, the coupling constants are resolved but not the indices. E.g. for the $H\ell_i\ell_j$ coupling in the electroweak Standard Model (Higgs–lepton–lepton with generation indices *i* and *j*, i.e. $\ell_1 = e, \ell_2 = \mu, \ell_3 = \tau$) they happen to be identical:

$$G_{H\ell_i\ell_j}^{\omega_-} = G_{H\ell_i\ell_j}^{\omega_+} = -\frac{ie}{2 \sin \theta_W M_W} \delta_{ij} m_{\ell_i}$$

Finally, at particles level, the generation indices *i* and *j* are resolved, so for instance the $He\mu$ coupling is zero because of the δ_{ij} .

The *FeynArts* functions `InsertFields`, `CreateFeynAmp`, and `Paint` can operate at different fields levels. The level specification is analogous to the usual *Mathematica* level specifications with e.g. `{3}` vs. `3`. If, for example, `InsertFields` is used with the option `InsertionLevel -> {Particles}`, the result will contain insertions *only at* particles level. In contrast, `InsertionLevel -> Particles` (no braces) will produce insertions *down to* particles level. More than one level may be specified, for instance, `{Generic, Particles}` skips the classes level.

4.2 InsertFields

The computer-algebraic generation of Feynman diagrams corresponds to the distribution of fields over topologies in such a way that the resulting diagrams contain only couplings allowed by the model. The function for this is `InsertFields`.

```
InsertFields[t, {i1, i2, ...} -> {o1, o2, ...}]
```

insert fields into the `TopologyList` *t* where the
incoming fields are *i₁, i₂, ...* and the outgoing
fields are *o₁, o₂, ...*

```
Create irreducible 1 → 2      In[9]:= t12 = CreateTopologies[1, 1 -> 2,
topologies.                  ExcludeTopologies -> Internal];
```

Insert $Z \rightarrow b\bar{b}$.

```
In[10]:= InsertFields[ t12,
               V[2] -> {F[4, {3}], -F[4, {3}]} ];


loading generic model file Models/Lorentz.gen
> $SVMixing is OFF
generic model Lorentz initialized

loading classes model file Models/SM.mod
> 46 particles (incl. antiparticles) in 16 classes
> $CounterTerms are ON
> 88 vertices
> 108 counter terms of order 1
> 1 counter terms of order 2
classes model SM initialized

inserting at level(s) {Generic, Classes}
> Top. 1: 6 Generic, 14 Classes insertions
> Top. 2: 0 Generic, 0 Classes insertions
> Top. 3: 0 Generic, 0 Classes insertions
> Top. 4: 0 Generic, 0 Classes insertions
in total: 6 Generic, 14 Classes insertions
```

Paint works also with
inserted topologies.

```
In[11]:= Paint[%, ColumnsXRows -> 4,  
          PaintLevel -> {Classes}]
```



InsertFields accepts the following options.

<i>option</i>	<i>default value</i>	
InsertionLevel	Classes	level specification (see end of Sect. 4.1)
GenericModel	"Lorentz"	generic model to use
Model	"SM"	classes model to use
ExcludeFieldPoints	{}	couplings to exclude
ExcludeParticles	{}	fields to exclude
Restrictions	{}	restrictions for diagram generation
LastSelections	{}	field patterns which must or must not appear in the final output

GenericModel specifies the model file containing the generic propagators and couplings (the extension .gen is always added to the file name). Model specifies the classes model

file containing the classes definitions and couplings (extension `.mod`). The model name is a string but may be given as a symbol if this is possible (e.g. a name like 2HD cannot be represented by a symbol because it starts with a digit).

4.3 Model Files

FeynArts distinguishes basic model files and add-on (or partial) model files. Most commonly, only the basic model files are used, as in

```
InsertFields[... , Model -> "MSSM"]
```

In contrast, the add-on model files do not supply a complete model. They just modify the particle descriptions and coupling tables of another model file and can therefore only be used “on top” of a basic model file. It produces an error to load an add-on model file without a basic one. An add-on model file might, for example, change a particular coupling, modify the mass of a particle, etc. One case given in [HaI06] is the enhancement (resummation) of the H - b - \bar{b} coupling in the MSSM. Such an add-on model file is used like

```
InsertFields[... , Model -> {"MSSM", "EnhHbb"}]
```

The model files that come with *FeynArts* are located in the `Models` subdirectory of the *FeynArts* tree. The major model files Standard Model, MSSM, and Two-Higgs-doublet Model are described in Appendices B, C, and D, respectively.

A classes or generic model file can also be initialized explicitly with `InitializeModel`. This can be useful e.g. when writing and debugging model files.

<code>InitializeModel[]</code>	initialize just the generic model file
<code>InitializeModel[<i>modname</i>]</code>	initialize the generic model file and the classes model file <i>modname.mod</i>

<i>option</i>	<i>default value</i>	
<code>GenericModel</code>	"Lorentz"	generic model to use
<code>Reinitialize</code>	True	whether to initialize the model, even if it is the current one
<code>ModelEdit</code>	Null	code that will be executed directly after loading the model

`GenericModel` specifies the model file containing the generic propagators and couplings (the extension `.gen` is always added to the file name). The same conventions for the model name as in the `Model` option of `InsertFields` apply here. `Reinitialize` specifies whether the model file is initialized when it is already the current model file.

`ModelEdit` provides a way to apply small changes to the model file, much as in the add-on model files described above. It provides code that is executed just after loading the model file, but before any initialization takes place. One example use could be the replacement of the ‘ordinary’ Z-boson mass by a complex one, which would include the width of the Z-boson. This could be done with

```
SetOptions[InitializeModel, ModelEdit :>
  (M$ClassesDescription = M$ClassesDescription /. MZ -> MZc)]
```

Note that the `ModelEdit` option uses `:>` (`RuleDelayed`) rather than `->` (`Rule`), otherwise the code would be executed immediately. For the description of the model-file contents, such as `M$ClassesDescription`, see Sect. 7.

4.4 Imposing Restrictions

It is often necessary to restrict the number of diagrams generated by `InsertFields`. This can be done in several ways.

`ExcludeParticles -> {fields}` excludes insertions containing *fields*. Patterns are permitted in *fields*, e.g. `ExcludeParticles -> F[_ , {2 | 3}]` excludes all second and third generation fermion fields. Note that excluding a field at a particular level automatically excludes derived fields at lower levels including their antiparticles. Excluding e.g. the classes field `F[1]` will also exclude `-F[1]`, `F[1, {1}]`, `-F[1, {1}]`, ... Furthermore, `ExcludeParticles` has no effect on the external particles. (After all, it would be rather pointless to specify certain external fields and exclude them in the same line.)

`ExcludeFieldPoints -> {couplings}` excludes insertions containing *couplings*. A coupling of counter-term order *cto* is specified as `FieldPoint[cto] [fields]`. Patterns are allowed, e.g. `FieldPoint[_] [-F[2, {1}], F[1, {1}], S[3]]`. Here, too, the exclusion of a field point entails the exclusion of more specific ones derived from it.

`Restrictions -> {exclp, exclfp}` is a convenient way of specifying abbreviations for `ExcludeParticles` and `ExcludeFieldPoints` statements. For example, `SM.mod` defines

```
NoElectronHCoupling =
```

```
ExcludeFieldPoints -> {
  FieldPoint[0][-F[2, {1}], F[1, {1}], S[3]],
  FieldPoint[0][-F[2, {1}], F[2, {1}], S[1]],
  FieldPoint[0][-F[2, {1}], F[2, {1}], S[2]] }
```

as a short-hand to exclude the electron–Higgs couplings. In `InsertFields` it is used as `Restrictions -> NoElectronHCoupling`.

`LastSelections` is an alternative method to specify field patterns that must or must not appear in the insertions. For example, `LastSelections -> {S, !F[_,{2}]}` forces that the insertions must contain a scalar field and must not contain fermions of the second generation. Like `ExcludeParticles`, `LastSelections` does not affect the external fields. The individual criteria are combined with the logical ‘and’, i.e. an insertion is only permitted if it fulfills all criteria simultaneously.


While `LastSelections` may seem more general, it works by first generating all diagrams and *afterwards* selecting those that match the given criteria (hence the name). In contrast, `ExcludeParticles` and `ExcludeFieldPoints` work by eliminating particles and couplings *before* starting the insertion process and can thus be significantly faster. They can, on the other hand, only exclude but not force the presence of fields or field points.

Here is the $Z \rightarrow b\bar{b}$ example again. Now we select only diagrams in which a W boson occurs.

```
In[12]:= InsertFields[ t12,
  V[2] -> {F[4, {3}], -F[4, {3}]},
  InsertionLevel -> {Classes},
  LastSelections -> V[3] ];
```

```
inserting at level(s) {Classes}
> Top. 1: 4 Classes insertions
> Top. 2: 0 Classes insertions
> Top. 3: 0 Classes insertions
> Top. 4: 0 Classes insertions
in total: 4 Classes insertions
```

```
In[13]:= Paint[%, ColumnsXRows -> 4]
```



4.5 Selecting Insertions


One further way to pick a selected set of diagrams is to use the function `DiagramSelect`.

`DiagramSelect[d, crit]` select the diagrams from *d* for which *crit* gives True

This function works much like the usual `Select` of *Mathematica*, i.e. it applies a test function to every diagram, and returns only those for which the result is `True`. The test function can of course be any *Mathematica* expression, but the most common usage is something like

```
DiagramSelect[ diags, FreeQ[#, Field[5] -> S]& ]
```

which eliminates all diagrams with a scalar particle on the fifth propagator. Such a statement would not be very useful were it not for the fact that *FeynArts* orders its propagators in a very systematic way. This is best exploited if the topologies are grouped into categories like self-energies, vertices, or boxes. For example, in a $2 \rightarrow 2$ vertex-correction diagram the first four propagators are those of the external particles, the fifth is the propagator of the tree part, and the rest are the loop propagators, e.g.



The important thing to note here is that the same “parts” of a diagram (say, the loop) in general get the same propagator numbers. Nevertheless, the number of a certain propagator is not fixed a priori, so the only reliable way to find the propagator numbers is to paint the bare topologies with the option `FieldNumbers -> True` (see Sect. 5).

The test function actually receives three arguments: the list of field substitution rules for the graph of the form `Graph[Field[1] -> f_1 , Field[2] -> f_2 , ...]`, the topology belonging to the graph, and the head of the surrounding topology list.

See Sect. 4.7 for auxiliary functions to use with `DiagramSelect`.

4.6 Grouping Insertions

The function `DiagramGrouping` groups insertions according to the output of a function.

<code>DiagramGrouping[<i>tops</i>, <i>foo</i>]</code>	return a list of parts of the inserted topologies <i>tops</i> , grouped according to the output of <i>foo</i>
---	---

The user function *foo* is applied to all insertions. Groups are introduced for all different return values of this function. The output of `DiagramGrouping` is a list of pairs (return value of *foo*) \rightarrow (inserted topologies).

The user function is invoked with the same three arguments as the test function of `DiagramSelect`: the list of field substitution rules for the graph of the form `Graph[Field[1] \rightarrow f_1 , Field[2] \rightarrow f_2 , ...]`, the topology belonging to the graph, and the head of the surrounding topology list.

See Sect. 4.7 for auxiliary functions to use with `DiagramGrouping`.

4.7 Auxiliary Functions

Several functions aid the selection of diagrams with `DiagramSelect` or their grouping with `DiagramGrouping`.

<code>LoopFields[<i>top</i>]</code>	return a list of the fields that are part of any loop in the topology <i>top</i>
<code>WFCorrectionFields[<i>top</i>]</code>	extract the fields external to any wave-function correction from topology <i>top</i>
<code>WFCorrectionCTFields[<i>top</i>]</code>	extract the fields external to any wave-function-correction counter-term from topology <i>top</i>
<code>LoopFields[<i>rul</i>, <i>top</i>]</code>	
<code>WFCorrectionFields[<i>rul</i>, <i>top</i>]</code>	
<code>WFCorrectionCTFields[<i>rul</i>, <i>top</i>]</code>	
	substitute the insertion rules <i>rul</i> into the bare topology <i>top</i> , then proceed as above

`LoopFields` identifies the fields running in the loop of a diagram. It is commonly used as in

```
DiagramSelect[ diags, FreeQ[LoopFields[##], V[1]]& ]
```

WFCorrection(CT)Fields typically returns a list of two fields, such as {S[1], S[3]}. These are the fields external to the wave-function correction (or its counter-term), i.e. the diagram contains a self-energy insertion $S[1] \rightarrow S[3]$ on an external leg. If the diagram contains no wave-function correction, the list is empty.

This filter is usually used to eliminate wave-function corrections with identical external legs, i.e. remove corrections of the type $a \rightarrow a$ but keep $a \rightarrow b$. This can be done with a construction like

```
DiagramSelect[ diags, UnsameQ@@ WFCorrectionFields[##] & ]
```

Vertices[<i>top</i>]	return the vertices contained in the topology <i>top</i>
FieldPoints[<i>top</i>]	return the field points contained in the topology <i>top</i>
FieldPoints[<i>rul</i> , <i>top</i>]	substitute the insertion rules <i>rul</i> into the bare topology <i>top</i> , then proceed as above

Vertices returns the vertices contained in a topology, not counting the external legs (even though they are internally represented as Vertex[1][*n*]).

FieldPoints returns the field content for each vertex of a topology, i.e. a list of objects of the form FieldPoint[*cto*][*fields*], where *cto* is the counter-term order.

The following functions further facilitate matching of fields and field points.

FieldMatchQ[<i>f</i> , <i>f'</i>]	True if the field <i>f</i> matches the pattern <i>f'</i>
FieldMemberQ[<i>flist</i> , <i>f'</i>]	True if an element of <i>flist</i> matches the field pattern <i>f'</i>
FieldPointMatchQ[<i>fp</i> , <i>fp'</i>]	True if the field point <i>fp</i> matches the pattern <i>fp'</i>
FieldPointMemberQ[<i>fpelist</i> , <i>fp'</i>]	True if an element of <i>fpelist</i> matches the field-point pattern <i>fp'</i>

FieldMatchQ works like MatchQ but takes into account field levels, e.g. F[1] matches F.

FieldPointMemberQ similarly works like MemberQ except that the field matching is done with FieldMatchQ.

FieldPointMatchQ and FieldPointMemberQ are to field points what FieldMatchQ and FieldMemberQ are to fields, respectively.

<code>FermionRouting[<i>top</i>]</code>	find out the permutation of external fermions as routed through the inserted topology <i>top</i>
<code>FermionRouting[<i>rul</i>, <i>top</i>]</code>	substitute the insertion rules <i>rul</i> into the bare topology <i>top</i> , then proceed as above

`FermionRouting` returns a list of integers of which every successive two denote the end-points of a fermion line in the diagram. This function is typically used as a filter for `DiagramSelect`, as in

```
DiagramSelect[ diags, FermionRouting[##] == {1, 4, 2, 3} & ]
```

or as

```
DiagramGrouping[ diags, FermionRouting ]
```

which returns a list of fermion-flow-ordered diagrams.

The function `FeynAmpCases` works like a `Cases` statement on the amplitude corresponding to a graph. That is, `FeynAmpCases` invokes `CreateFeynAmp` on each graph and from the resulting amplitude selects the parts matching a pattern.

<code>FeynAmpCases[<i>patt</i>] [<i>g</i>, <i>t</i>, <i>h</i>]</code>	create an amplitude from <i>g</i> , <i>t</i> , <i>h</i> (same three arguments as for user function above) and select all parts matching <i>patt</i> from that
<code>FeynAmpCases[<i>patt</i>] [<i>amp</i>]</code>	select all parts matching <i>patt</i> from the amplitude <i>amp</i>

This function is typically used as a filter for `DiagramSelect` or `DiagramGrouping`, as in

```
DiagramGrouping[ diags,
  FeynAmpCases[_[Index[Colour | Gluon, _], ___]] ]
```

4.8 Discarding Insertions

If the various ways of restricting `InsertFields` are not sufficient to select the desired diagrams, the user may discard the superfluous ones by number. The function for this is `Discard`.

<code>Discard[t, sel]</code>	discard the diagrams <i>sel</i> from the list of inserted topologies <i>t</i> . <i>sel</i> can be e.g. 5, 10...12, 20 which discards diagrams 5, 10, 11, 12, and 20.
------------------------------	--

The numbers referred to are the sequential numbers of the diagrams as given by `Paint`. Note that if `Paint` is used with a `PaintLevel` different from the `InsertionLevel`, the numbering will not be useful for `Discard`. Discarding a diagram at a particular level removes derived diagrams at deeper levels. For other kinds of objects, `Discard` works like an extended version of `Drop`, e.g. `Discard[{a, b, c, d, e}, 2...4]` results in `{a, e}`. `DiagramComplement` returns the complement of a list of diagrams.

<code>DiagramComplement[t_{all}, t₁, t₂, ...]</code>	the diagrams in <i>t_{all}</i> which are not in any of the <i>t_i</i>
--	---

4.9 Modifying Insertions

Diagrams can also be modified by mapping a function over the diagrams with `DiagramMap`.

<code>DiagramMap[foo, tops]</code>	map <i>foo</i> over each diagram in the inserted topologies <i>tops</i>
------------------------------------	---

The user function *foo* is applied to all insertions. It is invoked with the same three arguments as the test function of `DiagramSelect`: the list of field substitution rules for the graph of the form `Graph[Field[1] -> f1, Field[2] -> f2, ...]`, the topology belonging to the graph, and the head of the surrounding topology list. It must return the modified first argument (the `Graph` object).

4.10 Structure of the Inserted Topologies

If one wants to perform more advanced operations on inserted topologies it is necessary to know their structure. Essentially, a topology gets enclosed in a hierarchy of rules with particles insertions nested inside classes insertions nested inside generic insertions. It is possible to select specific levels out of this hierarchy with `PickLevel` (see Sect. 6.4).

```

TopologyList[info] [
  Topology[s] [props] ->
    Insertions[Generic] [
      Graph[sg1, Generic == 1] [Field[1] -> F, ...] ->
        Insertions[Classes] [
          Graph[sc1, Classes == 1] [Field[1] -> F[1], ...] ->
            Insertions[Particles] [
              Graph[sp1, Particles == 1] [Field[1] -> F[1, {1}], ...],
              Graph[sp2, Particles == 2] [Field[1] -> F[1, {2}], ...],
              more particles insertions
            ],
          Graph[sc2, Classes == 2] [Field[1] -> F[2], ...] -> ...,
          more classes insertions
        ],
      Graph[sg2, Generic == 2] [Field[1] -> V, ...] -> ...,
      more generic insertions
    ],
  more topologies
]

```

5 Drawing Feynman Graphs

The drawing routine `Paint` has already been used several times to illustrate the examples in this manual.

<code>Paint[t]</code> paint the (bare or inserted) topologies <i>t</i>
--

`Paint` accepts either a `TopologyList` or a `Topology` as argument. Note however that `InsertFields` adds an information field to `TopologyList` which is used by `Paint`. Thus, if you want to paint a single *inserted* topology, it is better to paint a topology list with one element (viz. `Take[toplist, {1}]`) than a single topology (viz. `toplist[[1]]`) in order to preserve that information field.

<i>option</i>	<i>default value</i>	
<code>PaintLevel</code>	<code>InsertionLevel</code>	level specification (see end of Sect. 4.1)
<code>ColumnsXRows</code>	3	number of diagrams per column/row, may also be given as a list $\{n_c, n_r\}$
<code>SheetHeader</code>	<code>Automatic</code>	title for each sheet of graphics
<code>Numbering</code>	<code>Full</code>	what type of numbering to display underneath each diagram
<code>FieldNumbers</code>	<code>False</code>	whether to label the propagators of a bare topology with the field numbers
<code>AutoEdit</code>	<code>True</code>	whether to call the topology editor when encountering an unshaped topology
<code>DisplayFunction</code>	<code>\$DisplayFunction</code>	which function to apply to the final graphics object in order to display it

`PaintLevel` specifies the level at which diagrams are painted. The default is to use the level the topologies were inserted at (for inserted topologies). For bare topologies `PaintLevel` is not relevant. Note that for the numbering of the diagrams to be appropriate for discarding insertions (see Sect. 4.8), the `PaintLevel` has to be the same as the `InsertionLevel`.

`ColumnsXRows` specifies the number of diagrams displayed in each column and row of a sheet. It may be given as a single integer, e.g. `ColumnsXRows -> 4` which means 4 rows

of 4 diagrams each on a sheet, or as a list of two integers, e.g. `ColumnsXRows -> {3, 5}` which means 5 rows of 3 diagrams each.

`SheetHeader` specifies the title of each sheet of diagrams. With `Automatic` or `True` the default header is used (“ $\text{particles}_{\text{in}} \rightarrow \text{particles}_{\text{out}}$ ” for inserted or “ $n_{\text{in}} \rightarrow n_{\text{out}}$ ” for bare topologies), `False` disables the header, and everything else is taken literally as a header, e.g. `SheetHeader -> "Self-energy diagrams"`.

`Numbering -> None` omits the default numbering of the diagrams. The default setting of `Numbering -> Full` places diagram numbers of the form T1 C8 N15 (= topology 1, classes insertion 8, running number 15) underneath the diagram. A simple numbering which is useful in publications is `Numbering -> Simple` in which case just the running number is used.

The propagators of a bare topology are usually unlabelled. With `FieldNumbers -> True` `Paint` uses the field numbers (the n in `Field[n]`) as labels. This is useful when selecting diagrams with `DiagramSelect` (see Sect. 4.5).

`AutoEdit` determines whether `Shape` is called when an unshaped topology is found. This option is useful if your diagrams involve lots of unshaped topologies (e.g. try painting 2-loop, $2 \rightarrow 4$ diagrams) and you want a quick (pre)view of the diagrams without going through the effort of shaping them at that moment. *FeynArts* uses a rather sophisticated autoshaping algorithm, but one should not expect too much: autoshaping offers a reasonable starting point, but for publishing-quality diagrams it is usually necessary to refine the shapes by hand.

`DisplayFunction` determines how the output of `Paint` is rendered. The default is to show the graphics on screen. Popular choices are `DisplayFunction -> Identity` which does not render the graphics at all, or `DisplayFunction -> (Display["file.ps", #]&)` which saves it in a file.

When running in the standalone *Mathematica* Kernel, `Paint` automatically animates the screen graphics, with the sheets of output acting as the frames of the animation. This is to inhibit the cluttering of the screen with too many windows. The speed at which the frames are displayed by default is rather high, more appropriate for a movie than for viewing sheets of graphics. The frame rate can be adjusted in the “Animation Controls” menu, which also allows to view single frames. Under Unix, the initial frame rate can be set with the X resource `Motifps*frameTime`, e.g. in `.Xdefaults`:

```
Motifps*frameTime: 1000
```

5.1 Things to do with the Paint Output

The output of `Paint` is a `FeynArtsGraphics` object whose arguments represent the sheets of the output. Each of these sheets is represented by a matrix which determines the layout of the diagrams, i.e.

```
FeynArtsGraphics[title] [
  { {g11, g12, ...},
    {g21, g22, ...},
    ... },
  ...
]
```

} sheet 1

These `FeynArtsGraphics` objects can be rendered with the usual *Mathematica* rendering functions `Export`, `Display`, and `Show`. `Export` and `Display` allow the image size to be changed with the `ImageSize` option whose default value is `72*{6, 7}` (6×7 inches) for a `FeynArtsGraphics` object.

In addition to the standard formats understood by *Mathematica*, the rendering functions accept two more output formats, "PS" and "TeX". To run output produced with the "TeX" format through \LaTeX , the `feynarts.sty` style file is required; it is located in the *FeynArts* directory.

One particular feature of the "TeX" output is that it can be touched up quite easily. This is useful for publications, e.g. when grouping together diagrams. To start with, the generated \LaTeX code doesn't look too scary, for instance a single diagram might be drawn by

```
\FADiagram{T1 C1 N2}
\FAProp(0.,15.)(6.,10.)(0.,){/Sine}{0}
\FALabel(2.48771,11.7893)[tr]{$Z$}
\FAProp(0.,5.)(6.,10.)(0.,){/Sine}{0}
\FALabel(3.51229,6.78926)[t1]{$Z$}
\FAProp(20.,15.)(14.,10.)(0.,){/Sine}{0}
\FALabel(16.4877,13.2107)[br]{$Z$}
\FAProp(20.,5.)(14.,10.)(0.,){/Sine}{0}
\FALabel(17.5123,8.21074)[b1]{$Z$}
\FAProp(6.,10.)(14.,10.)(0.,){/ScalarDash}{0}
```

```

\FALabel(10.,9.18)[t]{$H$}
\FAVert(6.,10.){0}
\FAVert(14.,10.){0}

```

With this representation, it is pretty straightforward to move diagrams around as they always start with `\FADiagram`. (Incidentally, the PostScript files generated by *FeynArts* have a very similar markup.)

The `feynarts.sty` introduces the following new \LaTeX commands:

<code>\begin{feynartspicture}</code>	
<code>\end{feynartspicture}</code>	delimit a sheet of diagrams
<code>\FADiagram{t}</code>	advance to the next diagram, which has title t
<code>\FAProp(f)(t)(c){g}{a}</code>	draw a propagator from point f to point t with curvature c using graphical representation g and arrow a
<code>\FAVert(p){cto}</code>	draw a vertex of counter-term order cto at point p
<code>\FALabel(p)[align]{text}</code>	write $text$ at point p with alignment $align$

Do not be confused by the multitude of parameters: The only command likely to be edited is `\FALabel`. Suffice it to say that each diagram is drawn on a 20×20 grid with the origin in the lower left corner and the positive axes extending to the right and upward, and that the alignment is specified in the standard \TeX manner, i.e. with combinations of `t`, `b`, `l`, and `r` for top, bottom, left, and right alignment, respectively. The exact details of all \LaTeX commands defined by `feynarts.sty` can be found in Appendix E.

5.2 Shaping Topologies


A topology does not by itself provide information on how to draw it. While the human eye is usually quite skilled in figuring out a nice shape for a topology, at least for not too complex ones, this is a tremendously difficult task for the computer even in simple cases. Indeed, the autoshaping routine is the longest single function in the whole program.

Whenever `Paint` encounters a topology for which it does not yet know the shape, it first does its best to autoshape the topology, and then calls the topology editor (unless `AutoEdit -> False` is set) so that the user can refine the shape. This shaping function

can also be invoked directly.

Shape [*t*] invoke the topology editor to shape the topology
or topology list *t*

The topology editor pops up a window which looks like



The *red squares* mark the vertices. Click and drag a red square with the left mouse button to move the respective vertex.

The *blue squares* mark the propagators. Click and drag a blue square with the left mouse button to change the respective propagator's curvature. Click on it with the middle mouse button to make the propagator straight again. Click on it with the right mouse button to reverse the curvature, i.e. to make the line curved in the opposite direction.

The *green squares* mark the label positions (they look like little tags attached to the center of the propagator). Click and drag a green square with the left mouse button to move the respective label. Click on it with the middle mouse button to put the label back to its

default position. Click on it with the right button to flick the label to the opposite side of the propagator.

The buttons on the right of the window are largely self-explanatory:

The OK button commits the changes and exits.

The CANCEL button discards the changes made to the current diagram and exits the editor but continues editing more diagrams if a list of topologies is being shaped.

The ABORT button is similar to CANCEL, but aborts the editing process altogether, i.e. it returns to the *Mathematica* prompt immediately, even if there are more topologies in line to be edited.

The REVERT button reverts to the initial layout of the topology.

Choosing GRID POSITION allows dragging of the squares only onto grid points. Conversely, ANY POSITION allows a square to be dragged to an arbitrary position.

The topologies need to be shaped only once. The shapes changed during a *Mathematica* session are saved in the directory specified by the variable `$ShapeDataDir`, which by default is the ShapeData directory in the *FeynArts* home directory.

The actual topology editor is a program written in Java, `TopologyEditor.java`, that communicates with *Mathematica* through the J/Link program. See Sect. 1 for setting up Java and J/Link.

6 Creating the Analytic Expressions

6.1 Representation of Feynman Amplitudes

The basic object for a Feynman amplitude is `FeynAmp`. Corresponding to the three insertion levels there are also three amplitude levels.

<i>for a single level:</i>	
<code>FeynAmp[n, mom, amp]</code>	Feynman amplitude with name <i>n</i> , integration momenta <i>mom</i> , and analytic expression <i>amp</i>
<i>for multiple levels:</i>	
<code>FeynAmp[n, mom, amp, ru]</code>	Feynman amplitude with name <i>n</i> , integration momenta <i>mom</i> , and generic analytic expression <i>amp</i> including replacement rules <i>ru</i> to obtain the other levels
<code>FeynAmpList[info][amps]</code>	list of <code>FeynAmps</code>
<code>GraphID[id]</code>	identifier of an amplitude
<code>Integral[q₁, q₂, ...]</code>	representation of the integration momenta

A `FeynAmp` containing a single level has only a single analytic expression *amp*: the amplitude at that level. In case more than one level is present, *amp* is the generic-level amplitude, which can be transformed into the individual amplitudes of the deeper levels by applying the replacement rules *ru* with `PickLevel` (see Sect. 6.4).

FeynArts always produces purely symbolic expressions for Feynman amplitudes, e.g. `PropagatorDenominator[p, m]` is used to denote $1/(p^2 - m^2)$, simply to prevent *Mathematica* from performing any automatic simplification on it. In fact, *FeynArts* does not attempt to simplify *anything* in the amplitude (e.g. evaluate traces) because this would limit its applicability to a certain class of theories.

Apart from the relatively few symbols *FeynArts* uses by itself in amplitudes, all symbols defined by the model can of course appear in the amplitude.

<code>Mass [f]</code>	mass of field f
<code>GaugeXi [f]</code>	gauge parameter of field f
<code>G [±1] [cto] [f] [kin]</code>	symmetric (+1) or antisymmetric (−1) generic coupling of counter-term order cto involving the fields f and corresponding to the kinematical object kin of the coupling vector (see Sect. 7 for details)
<code>FourMomentum [t, n]</code>	n th momentum of type t (Incoming, Outgoing, or Internal)
<code>PropagatorDenominator [p, m]</code>	symbolic expression for $1/(p^2 - m^2)$ where p is the momentum and m the mass of the propagator
<code>FeynAmpDenominator [...]</code>	collection of <code>PropagatorDenominator</code> s belonging to loops
<code>MatrixTrace [o₁, o₂, ...]</code>	closed fermion chain involving the noncommuting objects o_1, o_2, \dots
<code>FermionChain [o₁, o₂, ...]</code>	open fermion chain involving the noncommuting objects o_1, o_2, \dots
<code>VertexFunction [o] [f]</code>	generic vertex function of loop order o with adjoining fields f
<code>SumOver [i, r, ext]</code>	indicates that the amplitude it is multiplied with is to be summed in the index i over the range r ; if r is an integer, it represents the range $1 \dots r$. ext is the symbol <code>External</code> if the summation is over an index belonging to an external particle [†]
<code>IndexSum [expr, {i, range}]</code>	the sum of $expr$ over the index i ; <code>IndexSum</code> has the same syntax as <code>Sum</code> , but remains unevaluated

[†]It is necessary to distinguish sums over internal and external indices. For example, a squared diagram is calculated by squaring the diagram and then summing over the external indices. On the other hand, an internal index summation (say, for a quark loop) must be done for each diagram before squaring.

6.2 CreateFeynAmp

Once the possible combinations of fields have been determined by `InsertFields`, the Feynman rules must be applied to produce the actual amplitudes. The function for this is `CreateFeynAmp`.


<code>CreateFeynAmp[i]</code>	translate the inserted topologies <i>i</i> into Feynman amplitudes
-------------------------------	--

<i>option</i>	<i>default value</i>	
<code>AmplitudeLevel</code>	<code>InsertionLevel</code>	level specification (see end of Sect. 4.1)
<code>PreFactor</code>	$-i (2\pi)^4$ (-4LoopNumber)	overall factor of the amplitude
<code>Truncated</code>	<code>False</code>	whether to remove external wave functions
<code>GaugeRules</code>	<code>{_GaugeXi -> 1,</code> <code>MG0 -> MZ,</code> <code>MGp -> MW}</code>	rules to enforce a particular choice of gauge
<code>MomentumConservation</code>	<code>True</code>	whether to enforce momentum conservation at each vertex

The default for `AmplitudeLevel` is to use the same level the topologies were inserted at. `PreFactor` specifies the overall factor every diagram is multiplied with. It may contain the symbol `LoopNumber` which is substituted by the actual number of loops of a topology. If `Truncated -> True` is set, `CreateFeynAmp` applies the truncation rules defined in the model file (`M$TruncationRules`) to the final result. These rules should discard external wave functions, typically spinors or polarization vectors.

To be able to produce amplitudes in an arbitrary gauge, the model file must of course contain the full gauge dependence of the propagators and couplings. For example, `Lorentz.gen` and `SM.mod` contain the gauge-dependent propagators (`V`, `S`, `U`) and couplings (`SUU`) for an arbitrary R_ξ -gauge. However, most people prefer to work in the Feynman gauge which is the most convenient one for calculating radiative corrections. Therefore, the default `GaugeRules` set all gauge parameters to unity. On the other hand, if one wants to check e.g. gauge independence of an expression, `GaugeRules -> {}` will keep the gauge parameters untouched.

Create one diagram from
the photon self-energy:



Generate the amplitudes.

```
In[14]:= t11 = CreateTopologies[1, 1 -> 1][[3]];
In[15]:= AA = InsertFields[t11, V[1] -> V[1],
      ExcludeParticles -> {F, V, U}];

Excluding 3 Generic, 20 Classes, and 36 Particles fields
inserting at level(s) {Generic, Classes}
> Top. 1: 1 Generic, 1 Classes insertions
in total: 1 Generic, 1 Classes insertions

Restoring 3 Generic, 20 Classes, and 36 Particles fields

In[16]:= CreateFeynAmp[AA]

creating amplitudes at level(s) {Generic, Classes}
> Top. 1: 1 Generic, 1 Classes amplitudes
in total: 1 Generic, 1 Classes amplitudes

Out[16]= FeynAmpList[
  Model -> SM, GenericModel -> Lorentz,
  InsertionLevel -> Classes, Restrictions -> {},
  ExcludeParticles -> {F, V, U},
  ExcludeFieldPoints -> {}, LastSelections -> {},
  Process -> {{V[1], p1, 0}} -> {{V[1], k1, 0}}][
  FeynAmp[
    GraphID[Topology == 1, Generic == 1],
    Integral[q1],
    ( $\frac{1}{32}$  RelativeCF
      FeynAmpDenominator[ $\frac{1}{q_1^2 - \text{Mass}[S[\text{Gen3}]]^2}$ ,
         $\frac{1}{(-p_1 + q_1)^2 - \text{Mass}[S[\text{Gen4}]]^2}$ ]
      (p1 - 2 q1)[Lor1] (-p1 + 2 q1)[Lor2]
      ep[V[1], p1, Lor1] ep*[V[1], k1, Lor2]
      G(0)SSV[(Mom[1] - Mom[2])[KI1[3]]]
      G(0)SSV[(Mom[1] - Mom[2])[KI1[3]])] / Pi4,
      {Mass[S[Gen3]], Mass[S[Gen4]],
        G(0)SSV[(Mom[1] - Mom[2])[KI1[3]]],
        G(0)SSV[(Mom[1] - Mom[2])[KI1[3]]], RelativeCF} ->
      Insertions[Classes][{MW, MW, I EL, -I EL, 2}] ] ]
```

The CreateFeynAmp output is by default displayed on the screen in a special human-readable format. Its full internal representation is shown here.

```
In[17]:= InputForm[%]
Out[17]//InputForm=
FeynAmpList[Model -> "SM", GenericModel -> "Lorentz",
  InsertionLevel -> Classes, Restrictions -> {},
  ExcludeParticles -> {F, V, U},
  ExcludeFieldPoints -> {}, LastSelections -> {},
  Process -> {{V[1], FourMomentum[Incoming, 1], 0}} ->
    {{V[1], FourMomentum[Outgoing, 1], 0}}][
FeynAmp[
  GraphID[Topology == 1, Generic == 1],
  Integral[FourMomentum[Internal, 1]],
  (I/32*RelativeCF*
    FeynAmpDenominator[
      PropagatorDenominator[FourMomentum[Internal, 1],
        Mass[S[Index[Generic, 3]]],
      PropagatorDenominator[-FourMomentum[Incoming, 1] +
        FourMomentum[Internal, 1],
        Mass[S[Index[Generic, 4]]] ]*
      FourVector[FourMomentum[Incoming, 1] -
        2*FourMomentum[Internal, 1], Index[Lorentz, 1]]*
      FourVector[-FourMomentum[Incoming, 1] +
        2*FourMomentum[Internal, 1], Index[Lorentz, 2]]*
      PolarizationVector[V[1],
        FourMomentum[Incoming, 1], Index[Lorentz, 1]]*
      Conjugate[PolarizationVector][V[1],
        FourMomentum[Outgoing, 1], Index[Lorentz, 2]]*
      G[-1][0][-S[Index[Generic, 3]],
        -S[Index[Generic, 4]], V[1]][
        FourVector[Mom[1] - Mom[2], KI1[3]]]*
      G[-1][0][S[Index[Generic, 3]],
        S[Index[Generic, 4]], V[1]][
        FourVector[Mom[1] - Mom[2], KI1[3]]])/Pi^4,
    {Mass[S[Index[Generic, 3]]],
      Mass[S[Index[Generic, 4]]],
      G[-1][0][-S[Index[Generic, 3]],
        -S[Index[Generic, 4]], V[1]][
        FourVector[Mom[1] - Mom[2], KI1[3]]],
      G[-1][0][S[Index[Generic, 3]],
        S[Index[Generic, 4]], V[1]][
        FourVector[Mom[1] - Mom[2], KI1[3]]],
      RelativeCF} ->
    Insertions[Classes][{MW, MW, I*EL, -I*EL, 2} ] ]
```

6.3 Interpreting the Results

Although the analytical expression of an amplitude (the third element of a `FeynAmp`) may look complicated as a whole, the origin of the individual parts can easily be recounted. In the preceding example the respective terms have the following meaning:

$I/32$ and Pi^{-4} together are the product of all scalar factors from the (generic) couplings and propagators including the prefactor of the diagram. The fact that both terms are displayed apart in the output is a peculiarity of *Mathematica*.

The symbol `RelativeCF` stands for the *relative* combinatorical factor of a classes or particles amplitude with respect to the generic amplitude. It is inserted by the replacement rules for the deeper levels.

`FeynAmpDenominator` collects all denominators belonging to a loop, i.e. those whose momentum is not fixed by conservation of the external momenta. If the structure of the propagators is more intricate (e.g. gauge-boson propagators in a general gauge), a sum of several `FeynAmpDenominators` can appear in one amplitude.

$(p_1 - 2 q_1) [Lor1]$ and $(-p_1 + 2 q_1) [Lor2]$ are the kinematical objects—four-vectors in this case—that come from the generic coupling structure, the first from the left and the second from the right vertex. `Lor1` and `Lor2` are Lorentz indices. The external momenta are assigned such that the total momentum flows in through the incoming particles, and out through the outgoing particles, i.e. $\sum_{in} p_i - \sum_{out} p_i = 0$, not $\sum_{all} p_i = 0$.

$ep[V[1], p_1, Lor1]$ and $ep^*[V[1], k_1, Lor1]$ are the polarization vectors of the incoming and outgoing photon, the latter of which is complex conjugated.

$G_{SSV}^{(0)}[(Mom[1] - Mom[2]) [KI1[3]]]$ is the generic coupling constant of the scalar-scalar-vector coupling associated with the kinematical object $(Mom[1] - Mom[2]) [KI1[3]]$, a four-vector. (The SSV coupling happens to be proportional to only one kinematical object.) The superscript (0) refers to counter-term order 0. The two G 's (one for the left and one for the right vertex) look identical but are not the same internally because in the human-readable output format some indices are suppressed.

The replacement rules (the fourth element of a `FeynAmp`) particularize the unspecified quantities in the generic expression. For the single classes insertion of the example in the last section the following substitutions are made. (The full form of the generic expressions is written out below in small print to show that all substitutions are distinct.)

```
Mass[S[Gen3]] -> MW
full: Mass[S[Index[Generic, 3]]]
```

```

Mass[S[Gen4]] -> MW
full: Mass[S[Index[Generic,4]]]

G(0)SSV[(Mom[1]-Mom[2])[KI1[3]]] -> I*EL
full: G[-1][0][-S[Index[Generic,3]], -S[Index[Generic,4]], V[1]][FourVector[Mom[1]-Mom[2], KI1[3]]]

G(0)SSV[(Mom[1]-Mom[2])[KI1[3]]] -> -I*EL
full: G[-1][0][S[Index[Generic,3]], S[Index[Generic,4]], V[1]][FourVector[Mom[1]-Mom[2], KI1[3]]]

RelativeCF -> 2

```

6.4 Picking Levels

The replacement rules are not of a form which can directly be applied with *expr / . rules*. Instead, `PickLevel` has to be used to select a particular level.

<code>PickLevel[lev][amp]</code>	pick level <i>lev</i> from the amplitude <i>amp</i>
<code>PickLevel[levs][toplist]</code>	pick level(s) <i>levs</i> from the list of inserted topologies <i>toplist</i>

For the case of an amplitude, only one level may be specified. `PickLevel` replaces the generic amplitude and the rules with the amplitude at the selected level (i.e. the `FeynAmp` contains only three elements after `PickLevel`) and adds a running number of the form `Number == n` to the `GraphID`. Once a particular level has been selected, it is no longer possible to choose other levels (even deeper ones), e.g. it is not possible to first select classes level and subsequently particles level from a `FeynAmpList`.

With inserted topologies, `PickLevel` works somewhat differently. First, it preserves the structure of the inserted topologies. While in a `FeynAmp` the fourth element (the replacement rules) is deleted, an inserted topology is still of the form *topology* -> *insertions* after `PickLevel`. Second, more than one level may be picked, with one exception: the generic level cannot be removed because it is needed to create the generic amplitudes. However, the syntax differs from the usual level specification: `PickLevel[lev][t]` selects *only* level *lev* from *t* (apart from the generic level); `PickLevel[{leva, levb}]` selects levels *leva* and *levb*.

If `CreateFeynAmp` finds that amplitudes at only one level are requested (either directly by `AmplitudeLevel`, or if the topologies were inserted at only one level), it automatically calls `PickLevel` so that the output will contain only the amplitudes at the requested level and no replacement rules.

6.5 General Structure of the Amplitudes

The three-level concept has a big advantage when simplifying the amplitudes. Since the kinematical structure is already determined at the generic level, certain algebraic simplifications can be performed for the generic amplitude only and need not be done over and over for every insertion. For example, a fermionic trace needs to be calculated only for the generic amplitude, from which the expressions for the individual fermion classes or particles are then obtained by applying respectively the classes- or particles-level replacement rules.

Owing to fast computers and powerful software for evaluating amplitudes at one-loop level, however, this conceptually superior technique has not been used much so far. Nevertheless, for higher-loop calculations it will likely become an inevitable modus operandi given the number of diagrams already a two-loop calculation typically involves.

To be able to simplify the amplitudes at different levels it is essential to understand how they are structured. The structure is similar to that of inserted topologies, i.e. an expression embedded in a hierarchical list of insertion rules. The main difference is that the basic ingredient is not a topology but the generic amplitude.

Note that only amplitudes containing more than one level are structured in this way. If only one level is selected in `CreateFeynAmp` or with `PickLevel`, the `FeynAmps` contain only the analytic expression of the amplitude at the chosen level and no rules.

```
FeynAmpList[info] [
  FeynAmp[gname, mom, generic amp,
    {Mass[F[Index[Generic, 1]]], ...} ->
      Insertions[Classes] [
        {MLE, ...} ->
          Insertions[Particles] [
            {MLE[1], ...}, {MLE[2], ...}, more particles insertions
          ],
        {MQU, ...} -> ...,
        more classes insertions
      ],
  ],
  more amplitudes
]
```

6.6 On Fermion Chains and Signs

One of the most asked-for details of amplitude generation is how the signs of fermionic diagrams are determined.

FeynArts generally uses the methods from [De92]. That is basically, the fermion flow is fixed at generic level and so-called flip rules are applied if it turns out that in the classes or particles insertion the flow is reversed. Open fermion chains—the ones terminating on external fermions—are ordered opposite to the fermion flow, i.e. the leftmost field in the chain is an antifermion, $-F$.

For amplitudes containing anticommuting fields (fermions or ghosts) the following signs are added:

- a minus for every closed fermion or ghost loop and
- the signature of the permutation that brings the ordinal numbers of the external fermions as connected by the fermion chains into descending order (descending because the fermion chains are ordered opposite to the fermion flow).

Consider these diagrams:



Following the fermion chains opposite to the arrows, the external particles are connected as $\{2-1, 3-4\}$ in the left diagram, and as $\{3-1, 2-4\}$ in the right diagram. The number of permutations needed to bring $\{2\ 1\ 3\ 4\}$ into descending order $\{4\ 3\ 2\ 1\}$ is odd, but even for $\{3\ 1\ 2\ 4\}$. Hence the s -channel diagram gets an additional minus sign.

Note that the signature of the permutation is invariant under interchange of any two whole fermion chains because each fermion chain contributes an even number (two) of external fermions.

In renormalizable theories at most two fermion fields can appear at a vertex, and therefore the construction of the fermion chains is totally unambiguous. This is not so in other, non-renormalizable theories, for instance in the Fermi model. ***FeynArts* cannot correctly build the fermion chains if vertices involving more than two fermions appear** because this information is simply not available from the Feynman rules. In such

a case `$FermionLines = False` must be set and the fermion fields must carry an additional kinematic index (e.g. a Dirac index) with which it is *afterwards* possible (i.e. *FeynArts* does not do this) to find the right concatenation. Also, the above mentioned signs are not resolved in this case.

6.7 Specifying Momenta

`CreateFeynAmp` can be made to use a certain momentum on a certain propagator. This is done by editing the inserted topologies and appending a fourth argument to the propagators whose momenta one wants to fix. (A convenient way to do so is to assign the inserted topologies to a variable and use the *Mathematica* function `EditDefinition[var]`.) This fourth argument will become the momentum's name. For example, the momentum assigned to the propagator

```
Propagator[Internal][Vertex[3][101], Vertex[3][100], Field[4], Q1]
```

will be `FourMomentum[Internal, Q1]`. There are two rules to observe: Do not use an integer as a name to avoid conflict with the automatically-generated momenta, and do not change the direction of the propagator! If you need the momentum flowing opposite to the direction of the propagator, give it a minus sign, e.g. `-Q1`.

While `CreateFeynAmp` takes care to eliminate automatically-generated momenta before user-specified ones when working through momentum conservation, one can prohibit the elimination of any momentum with the option `MomentumConservation -> False`.

6.8 Compatibility with *FeynArts* 1

Many programs based on *FeynArts* still use the simpler conventions of version 1. For them to work with *FeynArts* 2.2, a function is provided to translate the `CreateFeynAmp` output into the old format.

`ToFA1Conventions[expr]` convert *expr* into the conventions of *FeynArts* 1

Note that `ToFA1Conventions` only renames some symbols. The results may thus not be 100% *FeynArts* 1 compatible since certain kinds of expressions could not be generated with *FeynArts* 1 at all. (That, after all, is why the conventions had to be changed.) In particular, amplitudes containing more than one level will probably not be very useful in programs made for processing *FeynArts* 1 output.

7 Definition of a New Model

7.1 The Generic Model

In the generic model file the following items are defined:

- the kinematic indices of the fields (`KinematicIndices`),
- the generic propagators (`M$GenericPropagators`),
- the generic coupling vectors (`M$GenericCouplings`),
- the fermion flipping rules (`M$FlippingRules`),
- the truncation rules (`M$TruncationRules`),
- some optional “final-touch” rules (`M$LastGenericRules`).

Apart from these required definitions, the generic model file is a perfectly ordinary *Mathematica* input file and may contain any number of additional statements. For example, `Lorentz.gen` includes `Format` directives with which the objects it introduces are displayed in a nicer form in the output.

Probably the best way to learn how to set up a generic model file is by going through one of the provided generic models, `Lorentz.gen` or `QED.gen`.

Kinematic Indices

A kinematic index is an index transported along a propagator. Due to the special property of renormalizable theories to possess vertices which join at most two fermion fields, spinor indices are not necessary because *FeynArts* can construct the fermion chains itself (cf. Sect. 6.6).

`KinematicIndices[f] = {i}` definition of the kinematic indices for field f

For example, in the usual representation of the Poincaré group, only the vector bosons carry a Lorentz index. `Lorentz.gen` thus defines the following kinematic indices.

```

KinematicIndices[ F ] = {};
KinematicIndices[ V ] = {Lorentz};
KinematicIndices[ S ] = {};
KinematicIndices[ SV ] = {Lorentz};
KinematicIndices[ U ] = {}

```

There seems to be a mismatch for the scalar–vector mixing field SV: it is declared to have a Lorentz index even though the scalar half has no index at all. However, `CreateFeynAmp` knows about this special case and discards the index on the scalar side when resolving a coupling.

Generic Propagators

A generic propagator defines the kinematical structure of a propagator. In the generic model file the generic propagators are contained in the list `M$GenericPropagators`. Inside this list each propagator is declared by an equation for `AnalyticalPropagator`.

```

M$GenericPropagators  name of the list of generic propagators
AnalyticalPropagator[t][f] == expr
                        definition of the propagator of type t (Internal
                        or External) for field f

```

The nomenclature perhaps needs explanation: An External “propagator” is what is traditionally called external wave function, e.g. a spinor in the case of a fermion field. Internal propagators are the propagators in the usual sense connecting two internal field points.

The simplest case are scalar fields which have no external wave function, and propagator

$$\langle S_i(k) | S_i(k) \rangle = \frac{i}{k^2 - \xi_{S_i} M_{S_i}^2}.$$

The corresponding statements in `Lorentz.gen` are

```

AnalyticalPropagator[External][ s S[i, mom] ] == 1,
AnalyticalPropagator[Internal][ s S[i, mom] ] ==
  I PropagatorDenominator[mom, Sqrt[GaugeXi[S[i]]] Mass[S[i]]]

```

When initializing the generic model, *FeynArts* transforms these equations into definitions and for this purpose augments the symbols like `s`, `i`, and `mom` by patterns. It is

important to realize that `CreateFeynAmp` appends the momentum and kinematic indices to the field while resolving the propagators and couplings. These additional elements do not appear in the final output, but must be taken into account in the definition of `AnalyticalPropagator`.

$f[i, mom, \{\mu_1, \dots\} \rightarrow \{\nu_1, \dots\}]$

maximal internal pattern layout of field f with classes and particles indices summarized in i , momentum mom , and kinematic indices μ_1, \dots on the left and ν_1, \dots on the right side of the propagator

For instance, a vector field `V[1]` might intermediately be extended by `CreateFeynAmp` to `V[1, FourMomentum[Internal, 1], {Index[Lorentz, 1]} -> {Index[Lorentz, 5]}]`.

The vector-boson propagator is defined accordingly:

```
AnalyticalPropagator[External][ s V[i, mom, {li2}] ] ==
  PolarizationVector[V[i], mom, li2],
AnalyticalPropagator[Internal][ s V[i, mom, {li1} -> {li2}] ] ==
  -I PropagatorDenominator[mom, Mass[V[i]]] *
    (MetricTensor[li1, li2] - (1 - GaugeXi[V[i]]) *
      FourVector[mom, li1] FourVector[mom, li2] *
      PropagatorDenominator[mom, Sqrt[GaugeXi[V[i]]] Mass[V[i]]])
```

The `s` stands for a possible prefactor (a sign for antiparticles or ± 2 for mixing fields), and is transformed to the pattern `s_` in the final definition. `i` summarizes the class index and all possible particles indices (final pattern: `i_`) and `mom` stands for the momentum (final pattern: `mom_`). For an internal propagator, `li1` and `li2` are the kinematic indices on the left and right side of the propagator, respectively (final patterns: `li1_` and `li2_`), and an external propagator uses only one set of kinematic indices, `li1`. The names of the patterns (`s`, `i`, `mom`, etc.) can of course be freely chosen, but must be unique. The most common errors in this respect are assignments of `i` or `j` prior to employing them as patterns.

Noncommuting objects appearing in the kinematic vector (typically elements from the Dirac algebra) must be wrapped in `NonCommutative` so that their order does not get destroyed by `CreateFeynAmp`. This makes sense only for fermion (and perhaps ghost)

fields since there is no well-defined order for other types of field. The external spinors are for example noncommutative objects:

```
AnalyticalPropagator[External][ s F[i, mom] ] ==
  NonCommutative[
    If[SelfConjugate[F[i]], MajoranaSpinor, DiracSpinor][
      -mom, Mass[F[i]] ] ]
```

Generic Couplings

Analogous to a generic propagator, a generic coupling defines the kinematical structure of a coupling. The generic couplings are contained in the generic model file in the list `M$GenericCouplings`. Inside this list each coupling is declared by an equation for `AnalyticalCoupling`. By convention, all fields in `AnalyticalCoupling` are incoming.

<code>M\$GenericCouplings</code>	name of the list of generic couplings
<code>AnalyticalCoupling[f₁, f₂, ...] == G[s][f₁, f₂, ...] . {k₁, k₂, ...}</code>	definition of the coupling of fields f ₁ , f ₂ , ... with coupling-constant vector G[s][f ₁ , f ₂ , ...] and kinematic vector {k ₁ , k ₂ , ...}
<code>G[±1][f₁, f₂, ...]</code>	generic coupling-constant vector, symmetric for +1 and antisymmetric for -1

The kinematically extended field structure of the fields must be taken into account similar to the case of generic propagators. Whereas however a field on an internal propagator has two sets of kinematic indices {li1} → {li2} for the left and right side each (provided the field carries kinematic indices at all), a field involved in a coupling obviously needs at most one set of indices.

For example, the quartic gauge-boson coupling is defined in `Lorentz.gen` as

```
AnalyticalCoupling[ s1 V[i, mom1, {li1}], s2 V[j, mom2, {li2}],
  s3 V[k, mom3, {li3}], s4 V[l, mom4, {li4}] ] ==
  G[1][s1 V[i], s2 V[j], s3 V[k], s4 V[l]] .
  { MetricTensor[li1, li2] MetricTensor[li3, li4],
    MetricTensor[li1, li3] MetricTensor[li2, li4],
    MetricTensor[li1, li4] MetricTensor[li3, li2] }
```

The generic coupling “constant” G refers actually to a vector of coupling constants, one for each element of the kinematic vector. The VVVV coupling would appear in a textbook as

$$\begin{aligned} C(V_\mu^i(p_1), V_\nu^j(p_2), V_\rho^k(p_3), V_\sigma^\ell(p_4)) &= G_{V_i V_j V_k V_\ell}^{(1)} g_{\mu\nu} g_{\rho\sigma} + G_{V_i V_j V_k V_\ell}^{(2)} g_{\mu\rho} g_{\nu\sigma} + G_{V_i V_j V_k V_\ell}^{(3)} g_{\mu\sigma} g_{\rho\nu} \\ &= \vec{G}_{V_i V_j V_k V_\ell} \cdot \begin{pmatrix} g_{\mu\nu} g_{\rho\sigma} \\ g_{\mu\rho} g_{\nu\sigma} \\ g_{\mu\sigma} g_{\rho\nu} \end{pmatrix}. \end{aligned}$$

When looking up a coupling, the fields of a vertex in general have to be permuted to fit the definition in the model file. For this to work it is necessary that the kinematic vector closes under permutations of fields of one type. In the example above this means that for any permutation of $\{\mu, \nu, \rho, \sigma\}$ (all four fields are of the same type here) one must end up with another (or the same) element of the kinematic vector, which is indeed the case.

There is one exception from this closure requirement: if a permutation of the fields yields either an element of the kinematic vector or the negative of an element of the kinematic vector, the coupling is said to be antisymmetric and specified as $G[-1]$. An example of this is the triple gauge-boson coupling in `Lorentz.gen`.

```
AnalyticalCoupling[ s1 V[i, mom1, {li1}], s2 V[j, mom2, {li2}],
  s3 V[k, mom3, {li3}] ] ==
G[-1][s1 V[i], s2 V[j], s3 V[k]] .
{ MetricTensor[li1, li2] FourVector[mom2 - mom1, li3] +
  MetricTensor[li2, li3] FourVector[mom3 - mom2, li1] +
  MetricTensor[li3, li1] FourVector[mom1 - mom3, li2] }
```

The kinematic vector has in this case only one element,

$$C(V_\mu^i(p_1), V_\nu^j(p_2), V_\rho^k(p_3)) = G_{V_i V_j V_k} (g_{\mu\nu}(p_2 - p_1)_\rho + g_{\nu\rho}(p_3 - p_2)_\mu + g_{\rho\mu}(p_1 - p_3)_\nu) .$$

Because the kinematic vector contains the differences of the momenta, a permutation of $\{(p_1, \mu), (p_2, \nu), (p_3, \rho)\}$ will produce either the kinematic vector itself or its negative.

Fermion Flipping Rules

Matters are more complicated for fermionic couplings. When two fermion fields are not in the order in which the coupling is defined in the model file, simple permutation

of the fields does not suffice, but in addition an operation called “flipping of fermionic couplings” has to be performed on the kinematic vector.

The algorithm behind this was developed in [De92] and its main advantage is that it can deal with fermion-number-violating vertices, such as appear in supersymmetric models. The idea is that instead of ordering the fermion propagators according to the given fermion flow, one *chooses* a direction for the fermion lines. (Obviously one cannot define the fermion flow properly as soon as fermion-number violating couplings are present, which is why the standard method breaks down in that case.) If later it turns out that the chosen direction is opposite to the actual fermion flow, one has to apply a so-called flipping rule. The flipping rule is nothing but a charge conjugation, i.e.

$$\Gamma \xrightarrow{\text{flip}} C\Gamma^T C^{-1}$$

where Γ is some product of Dirac matrices and C is the charge-conjugation matrix. The flipping rules for all possible objects that can appear in Γ have been worked out in [De92] and the only necessary substitution is

$$\gamma_\mu \omega_\pm \xrightarrow{\text{flip}} -\gamma_\mu \omega_\mp$$

where $\omega_\pm = (\mathbb{1} \pm \gamma_5)/2$ are the chirality projectors. This is more than just flipping the sign of the whole expression (which could be effected with $G[-1]$) since also the other chirality projector has to be taken.

A case where $\gamma_\mu \omega_\pm$ occurs is the FFV coupling in `Lorentz.gen`:

```
AnalyticalCoupling[ s1 F[i, mom1], s2 F[j, mom2],
  s3 V[k, mom3, {li3}] ] ==
G[-1][s1 F[i], s2 F[j], s3 V[k]] .
{ NonCommutative[DiracMatrix[li3], ChiralityProjector[-1]],
  NonCommutative[DiracMatrix[li3], ChiralityProjector[+1]] }
```

FeynArts expects the generic model file to define `M$FlippingRules`. These rules are applied when *FeynArts* needs to match a fermionic coupling but finds only the other fermion permutation in the model file. For example, in `Lorentz.gen` the following flip rules are defined:

```
M$FlippingRules =
  NonCommutative[dm:_DiracMatrix | _DiracSlash,
    ChiralityProjector[pm_]] ->
  -NonCommutative[dm, ChiralityProjector[-pm]]
```

Truncation Rules

Truncation rules are needed for removing external wave functions when the option `Truncated -> True` is used in `CreateFeynAmp`. The rules depend of course on the symbols that are used for the external wave functions and must hence be defined in the generic model file.

In `Lorentz.gen` there are spinors and polarization vectors to be dealt with since scalar particles have no external wave functions. The truncation rules are then

```
M$TruncationRules = {
  _PolarizationVector -> 1,
  _DiracSpinor -> 1,
  _MajoranaSpinor -> 1
}
```

“Final-Touch” Rules

The last operation `CreateFeynAmp` performs on a generic amplitude is to apply the model-dependent `M$LastGenericRules`. These rules are optional and have no particular purpose. In `Lorentz.gen`, for example, they complex-conjugate the outgoing polarization vectors.

```
M$LastGenericRules = {
  (* outgoing vector bosons: throw away signs of momenta *)
  PolarizationVector[p_, _ k:FourMomentum[Outgoing, _], li_] :>
    Conjugate[PolarizationVector][p, k, li]
}
```

This rule has an interesting side-aspect: note that only the head `PolarizationVector` is wrapped in `Conjugate`. This is because the `M$LastGenericRules` are applied with `//.` (replace repeatedly until the expression no longer changes). If the whole polarization vector (including arguments) were wrapped in `Conjugate`, the rules would apply anew in every pass and lead to an endless loop.

7.2 The Classes Model

All particles of a model are arranged in classes. A class is conceptually similar, but not identical, to a multiplet (the fields in one class need not belong to a representation of some group). For single-particle model definitions each particle lives in its own class.

The classes model defines the actual classes of fields in a particular model. It should not define or rely on any kinematic objects so that it can be used with different generic model files.

In the classes model file the following items are defined:

- the index range of all possible particles indices (`IndexRange`),
- a function that detects if a vertex violates quantum number conservation (`ViolatesQ`),
- the classes and their attributes (`M$ClassesDescription`),
- the classes coupling vectors (`M$CouplingMatrices`),
- some optional “final-touch” rules (`M$LastModelRules`).

Just as the generic model file, the classes model file is an ordinary *Mathematica* input file apart from the three required items and may contain additional definitions. For example, the abbreviations for the `Restrictions` option of `InsertFields` which typically depend on classes fields are usually defined in the classes model file.

Range of Particles Indices

It is necessary to declare the index range of every particles index.

```
IndexRange[Index[t]] = {p}    declares the index of type t to have the values p
```

For example, the `Generation` index in `SM.mod` which counts the fermion generations has the range

```
IndexRange[ Index[Generation] ] = {1, 2, 3}
```

If a particular index should not be “unfolded” at particles level (i.e. *FeynArts* should not generate an extra diagram for every value the index can take on), its range may be

wrapped in `NoUnfold`. For instance, one usually does not want eight diagrams generated and drawn for every gluon propagating; this can be effected with

```
IndexRange[ Index[Gluon] ] = NoUnfold[ Range[8] ]
```

As a consequence, fields with specific values for such indices have no effect in places like `ExcludeParticles`. This means that it is possible to exclude either all gluons or none, but not only the gluon with index 5.

Conservation of Quantum Numbers

When *FeynArts* generates diagrams with vertex-function placeholders (see Sect. 3.4), it eliminates illegal ones by checking whether the quantum numbers of the fields joining at the vertex are conserved. To this end, it calls the function `ViolatesQ` for every possible 1PI vertex function. `ViolatesQ` receives as arguments the quantum numbers of the involved fields (times -1 for antiparticles) and must return `True` if the vertex violates the conservation of those quantum numbers. In the most common case of charge-like (additive) quantum numbers, the condition is that the sum of all quantum numbers does not vanish at the vertex, which can be coded simply as

```
ViolatesQ[ q__ ] := Plus[q] != 0
```

However, this is not true for all kinds of quantum numbers (R-parity for example), which is why the definition of `ViolatesQ` is given in the model file.

Note the logic: all vertices for which `ViolatesQ` does not explicitly return `True` are accepted. It is no error if there is no definition for `ViolatesQ` since that simply means that all 1PI vertex functions are allowed.

Classes Attributes

The attributes of each class is defined by an equation in the list `M$ClassesDescription`.

<code>M\$ClassesDescription</code>	name of the list of classes descriptors
<code>class == {$a_I \rightarrow s_I, \dots$}</code>	descriptor for a class whose attribute a_I is s_I, \dots

Consider the up-type quark class in `SM.mod`.

```

F[3] == {
  SelfConjugate -> False,
  Indices -> {Index[Generation]},
  Mass -> MQU,
  QuantumNumbers -> 2/3 Charge,
  MatrixTraceFactor -> 3,
  PropagatorLabel -> ComposedChar["u", Index[Generation]],
  PropagatorType -> Straight,
  PropagatorArrow -> Forward }

```

The first two fields in the descriptor are necessary: `SelfConjugate` and `Indices`. The others are optional and have default values.

<i>option</i>	<i>default value</i>	
<code>SelfConjugate</code>	(required)	how the field behaves under charge conjugation
<code>Indices</code>	(required)	list of indices the class carries
<code>MixingPartners</code>	{ <i>field</i> , <i>field</i> }	for mixing propagators: left and right partners, e.g. {S[3], V[3]}
<code>Mass</code>	Mass[<i>field</i>]	symbol to denote the mass of the class members
<code>Mixture</code>	<i>field</i>	linear combination the field is composed of
<code>QuantumNumbers</code>	{}	quantum numbers the field carries
<code>MatrixTraceFactor</code>	1	for fermions: factor with which closed fermion loops will be multiplied
<code>InsertOnly</code>	All	types of propagators on which the field may be inserted
<code>PropagatorLabel</code>	" <i>field</i> "	label for the propagator
<code>PropagatorType</code>	Straight	line type for the propagator
<code>PropagatorArrow</code>	None	Forward, Backward, or None

An ordinary field also has default “mixing” partners, though trivial ones; e.g. `V[2]` has mixing partners {`V[2]`, `V[2]`}.

Mass \rightarrow *cmass* specifies a symbol for the mass of the class. It is possible to distinguish masses for different types of propagators – External, Internal, and Loop – by adding the propagator type as in

```
Mass[Loop]  $\rightarrow$  Mloop
```

Several such mass definitions can be given, where Mass alone acts as the default. If omitted, the expression Mass[*field*, *propagatortype*] remains in the output of CreateFeynAmp. The masses of particles-level fields will appear as *cmass*[*indices*] unless specific symbols are declared with the function TheMass. SM.mod for instance declares the following symbols for the individual up-type quark masses (which would otherwise be MQU[1], MQU[2], and MQU[3]):

```
TheMass[ F[3, {1}] ] = MU;
TheMass[ F[3, {2}] ] = MC;
TheMass[ F[3, {3}] ] = MT
```

Mixture gives the field’s composition as a linear combination of fields. Only mixtures with fields of the same type are allowed, e.g. scalars can only be composed of other scalars. An antiparticle has to be wrapped in Field, so that the minus sign is not taken as part of the coefficient in the linear combination. Using this method one can e.g. specify the couplings in terms of interaction eigenstates and linear-combine them to mass eigenstates. Unless the original (“interaction”) fields are themselves added to M\$ClassesDescription, they disappear completely from the model after initialization. An application where it is useful to keep original fields (including all vertices mixing original and rotated fields) is a rotation of higher order, with the rotated fields running on tree-level propagators and the unrotated ones in the loops.

QuantumNumbers specifies which quantum numbers the field carries. In principle, any *Mathematica* expression can be used as a quantum number, though the more common choices are multiples of a symbol, e.g. $\frac{2}{3}$ Charge. An important point is that the quantum number of the antiparticle is minus the quantum number of the particle. This is just what one wants in the case of additive quantum numbers like charge, since indeed if the up-quark has $\frac{2}{3}$ Charge, the anti-up-quark has $-\frac{2}{3}$ Charge. However, it is for instance not true that if a particle has R-parity 1, its antiparticle has R-parity -1 . In that case, one has to choose RParity[± 1] instead of \pm RParity and supply an appropriate ViolatesQ function.

`MatrixTraceFactor` is a way to compensate for traces over indices which are not explicitly specified. For example, in the electroweak Standard Model (without QCD) the interactions are colour-diagonal and hence the quarks need no colour index. Still, when computing the trace over a quark loop, colour accounts for a factor 3 which is provided by `MatrixTraceFactor` \rightarrow 3 in the quark class descriptors. `CreateFeynAmp` will complain if fields with different `MatrixTraceFactors` appear in the same loop.

`InsertOnly` specifies on which propagator types the field may be inserted, e.g. for a background field `InsertOnly` \rightarrow {Internal, External} is needed since it is not allowed in loops.

The line of a propagator is in general part of a circle. This includes straight lines which can be regarded as the degenerate case of infinite radius. The line type is given with `PropagatorType` which can take the values `Straight`, `Sine` (photon-like), `Cycles` (gluon-like), `ScalarDash`, or `GhostDash`.

The `PropagatorLabel` may contain a letter (e.g. "Z"), a \LaTeX symbol (e.g. " γ "), or a composite character (e.g. `ComposedChar["u", "-"]`). The double backslash is really a single character for *Mathematica* since the first \backslash is needed to escape the second \backslash . Inside the quotes only one character or \LaTeX symbol may appear. Multiple characters must be put in a list, e.g. {" γ ", "Z"}.

`ComposedChar[t, sub, sup, bar]` composite label t with subscript sub , superscript sup , and accent bar ; the arguments sub , sup , and bar are optional but their position is significant, e.g. `ComposedChar[t, Null, sup]` corresponds to a label with a superscript only

For example, `ComposedChar[" ϕ ", "i", "*", " \bar "]` will produce the label $\bar{\phi}_i^*$ (\LaTeX representation `\bar{\phi}_i^*`). As for `Mass`, it is possible to define labels specific to a particular propagator type (External, Internal, Loop), e.g.

```
PropagatorLabel[Loop] -> "X"
```

Indices can be used in a label in the form `Index[t]` where t is the type of the index, like in `ComposedChar[" ν ", Index[Generation]]`. The rendering of indices in the graphical output is subject to the function `Appearance`, if that is defined for the particular type of index. By default, the number of the index is displayed.

```

Appearance[Index[t, n_Integer]] := a
                                definition of the graphical appearance for indices
                                of type t
                                Alph[n]   the nth lowercase latin letter
                                UCAmph[n]  the nth uppercase latin letter
                                Greek[n]   the nth lowercase greek letter
                                UCGreek[n] the nth uppercase greek letter

```

The functions `Alph`, `UCAmph`, `Greek`, and `UCGreek` are useful for converting the index numbers into Latin or Greek letters, if this is desired. For example, the following line in `SM.mod` makes the fermion generation indices appear as i, j, \dots

```
Appearance[ Index[Generation, i_Integer] ] := Alph[i + 8]
```

Similarly, `Lorentz.gen` adds this definition to render Lorentz indices as μ, ν, \dots

```
Appearance[ Index[Lorentz, i_Integer] ] := Greek[i + 11]
```

Particular labels for the fields at particles level can be given by assigning values to the function `TheLabel`. If no specific label is defined for a particles-level field, the label of the associated class is used. E.g. the labels of the individual up-type quarks are declared in `SM.mod` by

```

TheLabel[ F[3, {1}] ] = "u";
TheLabel[ F[3, {2}] ] = "c";
TheLabel[ F[3, {3}] ] = "t"

```

Without these definitions, the up, charm, and top quarks would be labelled according to their classes label: u_1 , u_2 , and u_3 .

In the case of a mixing field, `PropagatorLabel` and `PropagatorType` may be set to a list of two items, one for the left half and one for the right half. For example, `SM.mod` contains the class for a γ - Z mixing field (it is commented out by default).

```

V[4] == {
  SelfConjugate -> True,
  Indices -> {},

```

```

Mass -> MAZ,
MixingPartners -> {V[1], V[2]},
PropagatorLabel -> {"\\gamma", "Z"},
PropagatorType -> Sine,
PropagatorArrow -> None }

```

Classes Coupling Vectors

By far the most diligent task is to enter the actual coupling vectors of the model, in particular if one endeavours to enter also the counter terms.

The coupling vectors are contained in a list `M$CouplingMatrices`. Each coupling vector is defined by an equation. The convention is that all fields at a coupling are incoming, as in the generic model file.

<code>M\$CouplingMatrices</code>	name of the list of coupling vectors
$C[f_1, f_2, \dots] == \{\{c_0^{(1)}, c_1^{(1)}, \dots\}, \{c_0^{(2)}, c_1^{(2)}, \dots\}, \dots\}$	definition of the coupling for the classes-level fields f_1, f_2, \dots where the lower index of the c 's refers to the counter-term order and the upper index to the component of the kinematic vector
$CC[f_1, f_2, \dots] == \dots$	the same, except that simultaneously the charge-conjugated vertex is defined
<code>ConjugateCoupling[f][c]</code>	defines how the charge-conjugate of the expression c is derived for the coupling of the fields f

The name `M$CouplingMatrices` is justified because the equations assign to each coupling a list of lists, or matrix (a coupling constant for each counter-term order for each component of the kinematic vector).

The coupling vector must of course have as many entries as in the kinematic vector it corresponds to. Recall the example of the quartic gauge-boson coupling from the last section. It has a kinematic vector with three entries, $(g_{\mu\nu}g_{\rho\sigma}, g_{\mu\rho}g_{\nu\sigma}, g_{\mu\sigma}g_{\rho\nu})$, so accordingly the coupling vector must also have three entries. An actual classes-level quartic gauge-boson coupling then looks like

```

C[ -V[3], V[3], V[1], V[1] ] == -I EL^2 *
{ {2, 4 dZe1 + 2 dZW1 + 2 dZAA1 - 2 CW/SW dZZA1},
  {-1, -2 dZe1 - dZW1 - dZAA1 + CW/SW dZZA1},
  {-1, -2 dZe1 - dZW1 - dZAA1 + CW/SW dZZA1} }

```

This is the $W^+ W^- \gamma \gamma$ coupling in `SM.mod`. On the right side there is a list with three entries corresponding to the three entries in the kinematic vector. Each component is again a list where the first element is the usual (counter-term order 0, or tree-level) coupling, the second the counter-term order 1 coupling, and so on. The overall constant $-I EL^2$ has been pulled out for clarity. (*Mathematica* automatically threads multiplications over lists, e.g. $x*\{a, b\}$ becomes $\{x*a, x*b\}$.)

It is possible to enter counter-terms only for a subset of couplings. The components of the coupling vector must however always be filled up to the same counter-term order.

In the special case where the kinematic vector has only one element, one level of lists may be omitted, so for example

```

C[ S[2], S[2], V[3], -V[3] ] == I EL^2/(2 SW^2) *
{ {1, 2 dZe1 - 2 dSW1/SW + dZW1 + dZchi1} }

```

is equivalent to

```

C[ S[2], S[2], V[3], -V[3] ] == I EL^2/(2 SW^2) *
{1, 2 dZe1 - 2 dSW1/SW + dZW1 + dZchi1}

```

The classes fields in $C[f_1, f_2, \dots]$ must be in the same generic order as the corresponding `AnalyticalCoupling` defined in the generic model file. This means that if an analytical coupling is defined in the order `FFS`, the classes coupling may not be given in the order `SFF`.

Fields with particles indices must include a pattern for every index in the coupling definition. For example, the $\overline{\nu}_{j_1} e_{j_2} W^+$ coupling is defined as

```

C[ -F[1, {j1}], F[2, {j2}], -V[3] ] ==
I EL/(Sqrt[2] SW) IndexDelta[j1, j2] *
{ {1, dZe1 - dSW1/SW + dZW1/2 +
  Conjugate[dZfL1[1, j1, j1]]/2 + dZfL1[2, j1, j1]/2},
  {0, 0} }

```


j_1 and j_2 stand for one index each (they are transformed to the patterns $j_{1_}$ and $j_{2_}$ internally during model initialization). For each index a class possesses a separate pattern has to be specified.

Index-diagonal terms are multiplied with $\text{IndexDelta}[j_1, j_2]$. If the whole coupling is proportional to IndexDelta as in the last example, `InsertFields` will use this information to fix indices as far as possible already in the insertion process. In contrast, `InsertFields` cannot constrain other indices if only part of the coupling is diagonal, such as in the first-order counter-term of the $\bar{\nu}_{j_1} \nu_{j_2} Z$ coupling.

```
C[ -F[1, {j1}], F[1, {j2}], V[2] ] == I EL *
{ {gL[1] IndexDelta[j1, j2],
  IndexDelta[j1, j2] (gL[1] dZZZ1/2 + dgL[1]) +
  gL[1] dZfL1cc[1, j1, j2]},
  {0, 0} }
```

The full index diagonality of the tree-level coupling in this case is of course respected as usual.

The conjugated coupling (the part of the Lagrangian usually abbreviated as “+ h.c.” in textbooks) can either be entered directly, or by using `CC` instead of `C` in the definition of the coupling; e.g. in the Standard Model one could define both the $\bar{u}dW^+$ and the $\bar{d}uW^-$ vertex, or define either one using `CC`. However, when using the `CC` method, one has to specify how the coupling vector of the conjugated coupling is derived from the original one. This is done by defining the function `ConjugateCoupling`. `ConjugateCoupling` is applied to all elements of the coupling vector and is in general different from a plain `Conjugate`, for instance the i from the exponent of the path integral (i.e. $i \int d^4x L$) must not be conjugated. The conjugation procedure can depend on the kinematic structure of the vertex, e.g. couplings involving a ∂_μ in configuration space usually get an additional minus which comes from the Fourier transformation. Using the available field information, i.e. the f in `ConjugateCoupling[f][c]`, the kinematic vector can be obtained with `KinematicVector[ToGeneric[f]]`.

A drawback is that `ConjugateCoupling` is a rather dumb function which has to be taught how to conjugate every symbol, so it is probably less work to enter the conjugated couplings directly if only a few are affected. Giving no definition for `ConjugateCoupling` is not an error: the amplitudes will then simply contain the symbol `ConjugateCoupling`.

“Final-Touch” Rules

Analogous to the `M$LastGenericRules` in the generic model, the classes model file allows to define optional `M$LastModelRules`. These rules are applied as the last operation before `CreateFeynAmp` returns its result. To avoid endless loops it must be kept in mind that the `M$LastModelRules` are applied with `//.` (replace repeatedly until expression no longer changes).

`SM.mod` currently does not define any `M$LastModelRules`.

7.3 Add-on Model Files

Add-on model files can be used with both generic and classes models. An add-on model file is different in that it does not define (overwrite) all model-file ingredients, but modifies them only. As an example, consider that one wants to introduce a resummed Yukawa coupling. This is most simply achieved by substituting the quark mass in the numerator of the coupling by a resummed quark mass, which could be achieved with the following add-one model file:

```
ResumCoup[ (c:C[_. _F, _. _F, _. _S] == rhs_ ] :=
  c == (rhs /. Mass[F[t_, {g_, ___}]] -> MfResummed[t, g]);
ResumCoup[ other_ ] = other;
M$CouplingMatrices = ResumCoup/@ M$CouplingMatrices
```

One can similarly build up a basic model file by loading an existing model file explicitly and then modifying the ingredients as above. This has the advantage that the user has to specify only the resulting model file’s name, not the combination of basic and add-on model file. The above add-on model file could thus be turned into a basic model file with

```
LoadModel["SM"];
ResumCoup[ (c:C[_. _F, _. _F, _. _S] == rhs_ ] :=
  c == (rhs /. Mass[F[t_, {g_, ___}]] -> MfResummed[t, g]);
ResumCoup[ other_ ] = other;
M$CouplingMatrices = ResumCoup/@ M$CouplingMatrices
```

<code>LoadModel[<i>modname</i>]</code>	load the model file <i>modname.mod</i>
<code>LoadModel[<i>modname</i>, <i>ext</i>]</code>	load the model file <i>modname.ext</i>

Finally, it is possible to save the model file presently in memory with the function `DumpModel`. This is useful, for example, if the modification of the basic model file by the add-on model files takes a lot of time, in which case one can dump the resulting model-file contents in a new model file which then loads faster. For example:

```
LoadModel[{"MSSMQCD", "FV", "EnhYuk", "HMix"}];
DumpModel["FullMSSM.mod"]
```

<code>DumpModel[<i>file</i>]</code>	save the classes-model-file variables presently in memory in the classes model file <i>file</i>
<code>DumpModel[<i>file</i>, <i>s</i>...]</code>	include the symbols <i>s</i> in the variables to be saved in <i>file</i>
<code>DumpGenericModel[<i>file</i>]</code>	save the generic-model-file variables presently in memory in the generic model file <i>file</i>
<code>DumpGenericModel[<i>file</i>, <i>s</i>...]</code>	include the symbols <i>s</i> in the variables to be saved in <i>file</i>

For easier debugging of add-on model files, *FeynArts* can list the couplings which have changed or were added/deleted. One can either wrap the model to be debugged in `ModelDebug` in the model specification or set the global variable `$ModelDebug`.

<code>ModelDebug[<i>mod</i>]</code>	report changes when initializing add-on model file <i>mod</i>
-------------------------------------	---

<i>variable</i>	<i>default value</i>	
<code>\$ModelDebug</code>	False	whether changes introduced by add-on model files will be reported as a model is initialized
<code>\$ModelDebugForm</code>	Short[#, 5]&	the output form for debugging output when model debugging is enabled

The global variable `$ModelDebug` determines whether changes introduced by add-on model files will be reported as a model is initialized. It can be set to `True`, in which case debugging output will be generated for all add-on model files, or to the name (or list of names) of the add-on model file(s) to be debugged. Alternately, debugging is turned on for all models wrapped in `ModelDebug` in the model specification.

7.4 Model-file Generation

FeynArts includes an add-on package called *ModelMaker* which can generate the Feynman rules for the classes model file from the Lagrangian. While entering the Lagrangian requires an effort comparable to typing in the Feynman rules directly, it is often nicer to have the Lagrangian available rather than only the Feynman rules. For example, in situations where particles mix to form mass eigenstates, the Lagrangian can be entered in terms of the gauge eigenstates, which is usually much simpler. It is straightforward in *Mathematica* to replace each gauge eigenstate with the appropriate linear combination of mass eigenstates and then derive the Feynman rules from the resulting expression.

Generating a model file with *ModelMaker* requires two things:

- A template model file must be made which contains everything except the Feynman rules, i.e. the definition of `M$CouplingMatrices` (see Sect. 7.2). *ModelMaker* reads the template model file (extension `.mmod`), inserts the `M$CouplingMatrices` definition, and writes out the `.mod` classes model file.
- The Lagrangian must be entered in a format where the fields are specially marked and the conventions of the generic model file are used for the kinematic quantities.

Assuming that the template model file is called `MODEL.mmod`, the procedure to generate `MODEL.mod` is the following:

```
<< Models`ModelMaker`
Lagrangian = (put definition of Lagrangian here);
frules = FeynmanRules[Lagrangian];
WriteModelFile[CouplingVector[frules], "MODEL.mmod"]
```

The *ModelMaker* functions used above are

<code>FeynmanRules[L]</code>	derives the Feynman rules from the Lagrangian L
<code>CouplingVector[r]</code>	extracts the coupling vector of each Feynman rule in r according to the corresponding kinematic vector in the currently initialized generic model file
<code>WriteModelFile[r, "t.mmod"]</code>	splices the Feynman rules generated with <code>FeynmanRules</code> and <code>CouplingVector</code> into the template model file $t.mmod$, writing the results to $t.mod$

The format for entering the Lagrangian is as follows: Fields are marked by the function `Field` whose first entry is the field name in the form in which it appears in the classes model file, e.g. `F[3, {g}]`. Only internal indices (i.e. particle-level indices) may appear in the field name itself. If it becomes necessary to refer to the momentum or kinematic indices of a field in the kinematic part of the coupling, two optional entries can be added to `Field` in the form `Field[f, mom, {μ, ...}]`. For example, the term in the Lagrangian corresponding to the electron–positron–photon vertex could be defined by

```
EL Field[-F[2, {g1}]] . DiracMatrix[mu] . Field[F[2, {g2}]] *
  Field[V[1], {mu}]
```

Note the dot product which joins the noncommuting objects. Names for the indices and momenta, such as `g1`, `g2`, and `mu` in the example above, can be chosen freely since they will not appear in the final Feynman rule (internal indices are eliminated via the functional derivative and kinematic quantities are replaced by generic objects for easier matching).

The kinematic quantities appearing in the coupling must be entered as in the generic model file. Two exceptions are allowed: a lone `DiracMatrix[μ]` is automatically split into $\gamma_\mu \omega_+ + \gamma_\mu \omega_-$, and `mom[μ]` may be used as a short-hand for `FourVector[mom, μ]`. As *ModelMaker* must know the kinematical quantities it might encounter, some of the conventions of `Lorentz.gen` have been hard-coded into *ModelMaker*. *ModelMaker* therefore may not work with generic model files other than `Lorentz.gen`.

The template model file is the same as the complete model file except that it does not contain the Feynman rules, i.e. it declares `IndexRange`, `M$ClassesDescription`, and `M$LastModelRules`, but not `M$CouplingMatrices` (see Sect. 7.2). The definition of

M\$CouplingMatrices, which in the complete model file contains the Feynman rules, instead has the form

```
M$CouplingMatrices = <* M$CouplingMatrices *>
```

When the template model file is processed by `WriteModelFile`, the text enclosed between `<*` and `*>` is scanned as *Mathematica* input and replaced by the resulting output, thereby inserting the Feynman rules. The output is written to a file with the first “m” stripped from the extension (`MODEL.mmmod` \rightarrow `MODEL.mod`), so the template model file is not overwritten or deleted.

A The Lorentz Formalism

`Lorentz.gen` contains the definitions of the generic propagators and couplings for a relativistic field theory with scalar, spinor, and vector fields transforming according to the usual representation of the Poincaré group. It introduces the following symbols.

<code>MetricTensor[μ, ν]</code>	the metric tensor $g_{\mu\nu}$
<code>DiracSpinor[p, m, i]</code>	the spinor of a Dirac fermion with momentum p , mass m , and particles indices i
<code>MajoranaSpinor[p, m, i]</code>	the spinor of a Majorana fermion with momentum p , mass m , and particles indices i
<code>PolarizationVector[v, p, μ]</code>	the polarization vector of the vector boson v with Lorentz index μ associated with momentum p
<code>DiracMatrix[μ]</code>	the Dirac matrix γ_μ
<code>DiracSlash[p]</code>	$\gamma_\mu p^\mu$
<code>ChiralityProjector[± 1]</code>	the chirality projectors $\omega_\pm = (1 \pm \gamma_5)/2$
<code>FourVector[p, μ]</code>	the four-vector p_μ

The four spinor states can actually all be represented by just one symbol depending on its position in a `FermionChain` and type of momentum:

```

FermionChain[ spinor[FourMomentum[Incoming,  $n$ ],  $m, i$ ], ... ] =  $\bar{v} \dots$ 
FermionChain[ spinor[FourMomentum[Outgoing,  $n$ ],  $m, i$ ], ... ] =  $\bar{u} \dots$ 
FermionChain[ ..., spinor[FourMomentum[Incoming,  $n$ ],  $m, i$ ] ] =  $\dots \cdot u$ 
FermionChain[ ..., spinor[FourMomentum[Outgoing,  $n$ ],  $m, i$ ] ] =  $\dots \cdot v$ 

```

where *spinor* is either `DiracSpinor` or `MajoranaSpinor`. Majorana spinors are used for fermions with attribute `SelfConjugate -> True`.

Polarization vectors associated with outgoing momenta are brought into the form `Conjugate[PolarizationVector][v, p, μ]` by the `M$LastGenericRules`.

The symbol `FourVector` defined by `Lorentz.gen` should not be confused with the *FeynArts* symbol `FourMomentum`. The latter represents a momentum flowing along propagator lines, and it is very likely that using it for different purposes would upset the internal routines. In contrast, `FourVector` is not modified by *FeynArts*.

The following generic couplings are defined in Lorentz.gen. Antisymmetric couplings are labelled by a subscript $-$, symmetric ones by a subscript $+$.

$$C(V_\mu, V_\nu, V_\rho, V_\sigma) = \vec{G}_{VVVV} \cdot \begin{pmatrix} g_{\mu\nu} g_{\rho\sigma} \\ g_{\mu\rho} g_{\nu\sigma} \\ g_{\mu\sigma} g_{\nu\rho} \end{pmatrix}_+ \quad (\text{VVVV})$$

$$C(V_\mu(k_1), V_\nu(k_2), V_\rho(k_3)) = \vec{G}_{VVV} \cdot (g_{\mu\nu}(k_2 - k_1)_\rho + g_{\nu\rho}(k_3 - k_2)_\mu + g_{\rho\mu}(k_1 - k_3)_\nu)_- \quad (\text{VVV})$$

$$C(S, S, S, S) = \vec{G}_{SSSS} \cdot (1)_+ \quad (\text{SSSS})$$

$$C(S, S, S) = \vec{G}_{SSS} \cdot (1)_+ \quad (\text{SSS})$$

$$C(S, S, V_\mu, V_\nu) = \vec{G}_{SSVV} \cdot (g_{\mu\nu})_+ \quad (\text{SSVV})$$

$$C(S(k_1), S(k_2), V_\mu) = \vec{G}_{SSV} \cdot (k_1 - k_2)_\mu_- \quad (\text{SSV})$$

$$C(S, V_\mu, V_\nu) = \vec{G}_{SVV} \cdot (g_{\mu\nu})_+ \quad (\text{SVV})$$

$$C(F, F, V_\mu) = \vec{G}_{FFV} \cdot \begin{pmatrix} \gamma_\mu \omega_- \\ \gamma_\mu \omega_+ \end{pmatrix}_+ \quad (\text{FFV})$$

$$C(F, F, S) = \vec{G}_{FFS} \cdot \begin{pmatrix} \omega_- \\ \omega_+ \end{pmatrix}_+ \quad (\text{FFS})$$

$$C(U(k_1), U(k_2), V_\mu) = \vec{G}_{UVV} \cdot \begin{pmatrix} k_{1\mu} \\ k_{2\mu} \end{pmatrix}_+ \quad (\text{UVV})$$

$$C(S, U, U) = \vec{G}_{SUU} \cdot (1)_+ \quad (\text{SUU})$$

$$C(V_\mu(k_1), V_\nu(k_2)) = \vec{G}_{VV} \cdot \begin{pmatrix} g_{\mu\nu}(k_1 k_2) \\ g_{\mu\nu} \\ k_{1\mu} k_{2\nu} \end{pmatrix}_+ \quad (\text{VV})$$

$$C(S(k_1), S(k_2)) = \vec{G}_{SS} \cdot \begin{pmatrix} k_1 k_2 \\ 1 \end{pmatrix}_+ \quad (\text{SS})$$

$$C(F(k_1), F(k_2)) = \vec{G}_{FF} \cdot \begin{pmatrix} k_1 \omega_- \\ k_2 \omega_+ \\ \omega_- \\ \omega_+ \end{pmatrix}_- \quad (\text{FF})$$

$$C(S(k_1), V_\mu(k_2)) = \vec{G}_{SV} \cdot \begin{pmatrix} k_{1\mu} \\ k_{2\mu} \end{pmatrix}_+ \quad (\text{SV})$$

B The Electroweak Standard Model

The file `SM.mod` contains the electroweak Standard Model including all counter-terms of first order in the conventions of [De93].

coupling constants and masses:

EL	electron charge (Thomson limit)
CW, SW	cosine and sine of the weak mixing angle
MW, MZ, MH	W, Z, Higgs masses
MG0, MGp	Goldstone masses
MLE [g]	mass of lepton of generation g
ME, MM, ML	e, μ, τ masses
MQU [g]	mass of up-type quark of generation g
MU, MC, MT	up, charm, top quark masses
MQD [g]	mass of down-type quark of generation g
MD, MS, MB	down, strange, bottom quark masses
CKM [g, g']	quark mixing matrix
GaugeXi [A, W, Z]	photon, W, Z gauge parameters

one-loop renormalization constants:

dZe1	electromagnetic charge RC
dSW1, dCW1	mixing angle sine/cosine RC
dZH1, dMHsq1	Higgs field and mass RC
dZW1, dMWsq1	W field and mass RC
dMZsq1	Z mass RC
dZZZ1, dZZA1, dZAZ1, dZAA1	Z and photon field RCs
dZG01, dZGp1	Goldstone field RCs
dMf1 [t, g]	fermion mass RCs
dZfL1 [t, g, g'], dZfR1 [t, g, g']	left- and right-handed fermion field RCs
dCKM1 [g, g']	quark mixing matrix RCs

The type of a fermion is 1 for neutrinos, 2 for massive leptons, 3 for up-type quarks, and 4 for down-type quarks.

The particle content of `SM.mod` is summarized in the following table.

<i>class</i>	<i>self-conj.</i>	<i>indices</i>	<i>members</i>		<i>mass</i>
F[1] (neutrinos)	no	Generation	F[1, {1}]	ν_e	0
			F[1, {2}]	ν_μ	0
			F[1, {3}]	ν_τ	0
F[2] (massive leptons)	no	Generation	F[2, {1}]	e	ME
			F[2, {2}]	μ	MM
			F[2, {3}]	τ	ML
F[3] (up-type quarks)	no	Generation	F[3, {1, <i>o</i> }]	u	MU
		Colour	F[3, {2, <i>o</i> }]	c	MC
			F[3, {3, <i>o</i> }]	t	MT
F[4] (down-type quarks)	no	Generation	F[4, {1, <i>o</i> }]	d	MD
		Colour	F[4, {2, <i>o</i> }]	s	MS
			F[4, {3, <i>o</i> }]	b	MB
V[1]	yes		V[1]	γ	0
V[2]	yes		V[2]	Z	MZ
V[3]	no		V[3]	W^-	MW
V[4] (mixing field) [†]	yes		V[4]	$\gamma-Z$	MAZ
S[1]	yes		S[1]	H	MH
S[2]	yes		S[2]	G^0	MGO
S[3]	no		S[3]	G^-	MGp
U[1]	no		U[1]	u_γ	0
U[2]	no		U[2]	u_Z	MZ
U[3]	no		U[3]	u_-	MW
U[4]	no		U[4]	u_+	MW
SV[2] (mixing field) [‡]	yes		SV[2]	G^0-Z	MZ
SV[3] (mixing field) [‡]	no		SV[3]	G^-W^-	MW

[†]Commented out by default in `SM.mod`.

[‡]Must be enabled with `$SVMixing = True`.

SM.mod defines the following Restrictions:

NoGeneration1	exclude generation-1 fermions (ν_e, e, u, d)
NoGeneration2	exclude generation-2 fermions (ν_μ, μ, c, s)
NoGeneration3	exclude generation-3 fermions (ν_τ, τ, t, b)
NoElectronHCoupling	exclude all couplings involving electrons and a Higgs ($e^- e^+ H, e^- e^+ G^0, e^- \nu_e G^-$).
NoLightFHCoupling	exclude all couplings between light fermions (all fermions except the top) and Higgs fields ($f_i \bar{f}_i H, f_i \bar{f}_i G^0, \ell_i^- \nu_i G^-, \bar{d}_i u_i G^-, f_i \neq t, u_i \neq t$).
NoQuarkMixing	exclude all couplings where off-diagonal elements of the quark mixing matrix appear ($\bar{d}_i u_j W^-, \bar{d}_i u_j G^-, i \neq j$). Note that the diagonal elements $\text{CKM}[i, i]$ are nevertheless present.
QEDOnly	exclude all particles except the massive fermions, the photon, and the photon ghost.

SMew.mod is a companion model file for SM.mod in which the quarks do not carry colour indices. It is included for backward compatibility with old versions.

B.1 The QCD Extension

The model file SMQCD.mod adds the QCD Feynman rules to the electroweak part in SM.mod. In fact, it just loads SM.mod and appends the gluon and its ghost to the appropriate definitions.

Following is the list of additional symbols used in SMQCD.mod.

GS	the strong coupling constant
SUNT[a, i, j]	the generators of SU(N), $(T^a)_{ij}$
SUNF[a, b, c]	the SU(N) structure constants f^{abc}
SUNF[a, b, c, d]	a short-hand for the sum $\sum_i f^{abi} f^{icd}$

The additional particles are the gluon and its ghost. The gluon index is not expanded out at particles level, i.e. *FeynArts* does not generate eight diagrams for every gluon.

<i>class</i>	<i>self-conj.</i>	<i>indices</i>	<i>members</i>	<i>mass</i>
V[5]	yes	Gluon	$V[5, \{i\}] \quad g_i$	0
U[5]	no	Gluon	$U[5, \{i\}] \quad u_{g_i}$	0

B.2 Background-field Formalism

The model file `SMbgf.mod` contains the electroweak Standard Model in the background-field formalism. It is based on [De95], with the exception that the renormalization of the fermionic fields proceeds as in `SM.mod`, i.e. with separate renormalization constants for upper and lower components of the fermion doublet, which in turn means that fermions do not need an external wave-function renormalization. The larger part of `SMbgf.mod` is in fact derived from `SM.mod`. At present, vertices containing quantum fields other than fermions do not possess counter-terms, but this is sufficient for one-loop calculations.

To work with `SMbgf.mod`, one needs to specify `Lorentzbgf.gen` as the generic model, which is a slightly generalized version of `Lorentz.gen`.

The following table lists the background fields and their quantum-field counterparts. All other fields are the same as in `SM.mod`.

background field		quantum field	
V[10]	$\hat{\gamma}$	V[1]	γ
V[20]	\hat{Z}	V[2]	Z
V[30]	\hat{W}^-	V[3]	W^-
S[10]	\hat{H}	S[1]	H
S[20]	\hat{G}^0	S[2]	G^0
S[30]	\hat{G}^-	S[3]	G^-
SV[20] [§]	$\hat{G}-\hat{Z}$	SV[2] [§]	$G-Z$
SV[30] [§]	$\hat{G}^- - \hat{W}^-$	SV[3] [§]	$G^- - W^-$

[§]Must be enabled with `$SVMixing = True`.

C The Minimal Supersymmetric Standard Model

The file `MSSMQCD.mod` defines the complete (electroweak and strong) MSSM, whereas `MSSM.mod` contains only the electroweak subset, defined as everything except the gluon, its ghost, and the gluino [HaS02]. The four-sfermion couplings appear in `MSSM.mod` even though they have both electroweak and strong parts. Both model files follow the conventions of [Ha85, Gu86, HHG90]. These conventions differ from the ones in `SM.mod` by the sign of the $SU(2)$ covariant derivative. Counter-terms are currently not included in the MSSM model files.

The symbols used for the MSSM parameters are listed in the following table. The Standard-Model parameters have the same names as in `SM.mod`, e.g. `MW`, and are omitted from this table.

<code>Mh0, MHH, MA0, MGO</code>	the neutral Higgs boson masses
<code>MHp, MGp</code>	the charged Higgs boson masses
<code>CB, SB, TB</code>	$\cos \beta, \sin \beta, \tan \beta$
<code>CA, SA</code>	$\cos \alpha, \sin \alpha$
<code>C2A, S2A, C2B, S2B</code>	$\cos 2\alpha, \sin 2\alpha, \cos 2\beta, \sin 2\beta$
<code>CAB, SAB, CBA, SBA</code>	$\cos(\alpha + \beta), \sin(\alpha + \beta), \cos(\beta - \alpha), \sin(\beta - \alpha)$
<code>MUE</code>	the Higgs-doublet mixing parameter μ
<code>MG1</code>	the gluino mass
<code>MNeu[n]</code>	the neutralino masses
<code>ZNeu[n, n']</code>	the neutralino mixing matrix
<code>MCha[c]</code>	the chargino masses
<code>UCha[c, c'], VCha[c, c']</code>	the chargino mixing matrices
<code>MSf[s, t, g]</code>	the sfermion masses
<code>USf[t, g][s, s']</code>	the sfermion mixing matrix
<code>Af[t, g]</code>	the (scalar) soft-breaking A-parameters

The sfermion type is denoted by t and is defined similar to the fermion type. The primed indices appearing in the mixing matrices enumerate the gauge eigenstates, the unprimed ones mass eigenstates. The following indices are used in `MSSM.mod` and

MSSMQCD.mod:

$$\begin{aligned}
 g &= \text{Index}[\text{Generation}] = 1 \dots 3, & s &= \text{Index}[\text{Sfermion}] = 1 \dots 2, \\
 o &= \text{Index}[\text{Colour}] = 1 \dots 3, & n &= \text{Index}[\text{Neutralino}] = 1 \dots 4, \\
 u &= \text{Index}[\text{Gluon}] = 1 \dots 8, & c &= \text{Index}[\text{Chargino}] = 1 \dots 2.
 \end{aligned}$$

The particle content of MSSM.mod and MSSMQCD.mod is given in the next table. The gluon, its ghost, and the gluino, which are defined only in the latter, are written in grey. “sc” is short for self-conjugate.

leptons				mass	sleptons				mass
ν_g		F[1, {g}]		0	$\tilde{\nu}_g$		S[11, {g}]		MSf
ℓ_g		F[2, {g}]		MLE	$\tilde{\ell}_g^s$		S[12, {s, g}]		MSf

quarks					squarks				
u_g		F[3, {g, o}]		MQU	\tilde{u}_g^s		S[13, {s, g, o}]		MSf
d_g		F[4, {g, o}]		MQD	\tilde{d}_g^s		S[14, {s, g, o}]		MSf

gauge bosons					neutralinos, charginos				
γ	sc	V[1]		0	$\tilde{\chi}_n^0$	sc	F[11, {n}]		MNeu
Z	sc	V[2]		MZ	$\tilde{\chi}_c^-$		F[12, {c}]		MCha
W^-		V[3]		MW					

Higgs bosons					ghosts				
h^0	sc	S[1]		Mh0	u_γ		U[1]		0
H^0	sc	S[2]		MHH	u_Z		U[2]		MZ
A^0	sc	S[3]		MA0	u_+		U[3]		MW
G^0	sc	S[4]		MGO	u_-		U[4]		MW
H^-		S[5]		MHp	u_{g_i}		U[5, {i}]		0
G^-		S[6]		MGp					

gluon					gluino				
g_i	sc	V[5, {i}]		0	\tilde{g}_i	sc	F[15, {i}]		MGl

`MSSM.mod` and `MSSMQCD.mod` define the following Restrictions:

<code>NoGeneration1</code>	exclude generation-1 fermions (ν_e, e, u, d)
<code>NoGeneration2</code>	exclude generation-2 fermions (ν_μ, μ, c, s)
<code>NoGeneration3</code>	exclude generation-3 fermions (ν_τ, τ, t, b)
<code>NoElectronHCoupling</code>	exclude all couplings involving electrons and any Higgs field
<code>NoLightFHCoupling</code>	exclude all couplings between light fermions (all fermions except the top) and any Higgs field
<code>NoSUSYParticles</code>	exclude the particles not present in the SM: sfermions, charginos, neutralinos, and the Higgs fields H^0, A^0, H^\pm
<code>THDMParticles</code>	exclude the particles not present in the Two-Higgs-Doublet Model: the sfermions, charginos, and neutralinos

The complete list of all couplings defined in `MSSM.mod` and `MSSMQCD.mod` is contained in the files `MSSM.ps.gz` and `MSSMQCD.ps.gz`, respectively, which are located in the `Models` directory. The “couplings” in these PostScript files are actually the coupling vectors corresponding to the kinematic vectors defined in `Lorentz.gen`.

`MSSM.mod` and `MSSMQCD.mod` currently do not contain any counter-term vertices, so that counter-term diagrams cannot be generated automatically yet. This is due to the fact that, although it is in principle known how to renormalize theories with softly-broken supersymmetry [Ho00], this is far from trivial for the MSSM and has so far not been worked out completely. As long as only SM particles appear at tree-level, however, one can almost directly use the SM counter-terms. Of course the self-energies from which the renormalization constants are derived now have to be calculated in the MSSM.

The one thing one has to observe when using the SM counter-terms for an MSSM process is that, for historical reasons, the SM and MSSM model files differ in the sign of the $SU(2)$ covariant derivative. Namely,

$$D_\mu = \partial_\mu + \sigma \, ig_2 W_\mu,$$

where σ is $-$ in the SM and $+$ in the MSSM model files. There is a simple rule for translating the two conventions: replace SW by $-SW$ and add an additional minus sign for each Higgs field that appears in a coupling.

D The Two-Higgs-Doublet Model

The Two-Higgs-Doublet Model (THDM) possesses a Higgs sector similar to the MSSM (i.e. 5 physical Higgs fields), but has no supersymmetry and hence no superpartners. The Feynman rules are mostly the same as for the MSSM with the couplings involving superpartners omitted, except for the SFF, SSS, and SSSS couplings. Model constants are given in the following table. Again, the Standard-Model parameters are omitted since they are the same as in `SM.mod`.

Mh0, MHH, MA0, MG0	the neutral Higgs boson masses
MHp, MGp	the charged Higgs boson masses
CB, SB, TB	$\cos \beta, \sin \beta, \tan \beta$
CA, SA	$\cos \alpha, \sin \alpha$
C2A, S2A, C2B, S2B	$\cos 2\alpha, \sin 2\alpha, \cos 2\beta, \sin 2\beta$
CAB, SAB, CBA, SBA	$\cos(\alpha + \beta), \sin(\alpha + \beta), \cos(\beta - \alpha), \sin(\beta - \alpha)$
Yuk1, Yuk2, Yuk3	Yukawa-coupling parameters (see below)

One distinguishes two types of THDM, Type I, where all fermions couple only to the second Higgs doublet H_2 , and Type II, where up-type fermions couple to H_2 whereas down-type fermions couple to H_1 . In the *FeynArts* implementation of the THDM, this choice is parameterized by three Yukawa-coupling parameters which take the values

	Yuk1	Yuk2	Yuk3
Type-I THDM	$\frac{\cos \alpha}{\sin \beta}$	$\frac{\sin \alpha}{\sin \beta}$	$-\cot \beta$
Type-II THDM	$-\frac{\sin \alpha}{\cos \beta}$	$\frac{\cos \alpha}{\cos \beta}$	$\tan \beta$

The following Restrictions are defined in `THDM.mod`:

NoGeneration n	exclude generation- n fermions ($n = 1, 2, 3$)
NoElectronHCoupling	exclude all couplings involving electrons and any Higgs field
NoLightFHCoupling	exclude all couplings between light fermions (all fermions except the top) and any Higgs field

E Graphics Primitives in `feynarts.sty`

The *FeynArts* style is included in a \LaTeX 2 ϵ document with

```
\usepackage{feynarts}
```

It makes three graphics primitives available with which Feynman diagrams can be drawn:

- `\FAProp` draws a propagator,
- `\FAVert` draws a vertex,
- `\FALabel` places a label.

In addition, it provides formatting/geometry directives:

- `\begin...end{feynartspicture}` delineates a sheet of Feynman diagrams,
- `\FADiagram` advances to the next diagram.

Since `feynarts.sty` emits direct PostScript primitives, the interpretation of which is non-standard across PostScript renderers, it is guaranteed to work only with `dvips`.

E.1 Geometry


A single Feynman diagram is always drawn on a 20×20 canvas. Several such canvasses are combined into a rectangular sheet which can optionally carry a title. A sheet of Feynman diagrams is enclosed in a `feynartspicture` environment in \LaTeX :

```
\begin{feynartspicture}(s_x,s_y)(n_x,n_y)
...
\end{feynartspicture}
```

This sheet has a size of $s_x \times s_y$ (in units of \LaTeX 's `\unitlength`) with room for $n_x \times n_y$ Feynman diagrams. n_y need not be an integer and the extra space implied by the fractional part is allocated at the top for the sheet label.

Note that it is not possible to distort the aspect ratio of a Feynman diagram. If the ratio $n_x/\lfloor n_y \rfloor$ is chosen different from the ratio s_x/s_y , the sheet will fit the smaller dimension exactly and be centered in the larger dimension.

The overall geometry of a feynartspicture sheet is thus as follows (shown here for a 2×2 sheet):



Inside the feynartspicture, the macro

```
\FADiagram{dtitle}
```

advances to the next diagram, which has the title *dtitle*. The size of *dtitle* can be changed by redefining `\FADiagramLabelSize` with one of the usual L^AT_EX font-size specifiers, e.g.


```
\def\FADiagramLabelSize{\scriptsize}
```

The default size is `\small`.


E.2 Propagators

All propagators are circular arcs in the *FeynArts* style. This includes conceptually the straight line as the infinite-radius limit. Propagators furthermore come in two variants:

tadpole propagators, where the initial and final vertex coincide, and ‘ordinary’ propagators with distinct initial and final vertex. This distinction is necessary because the information that has to be stored is different for the two cases. The arguments of the `\FAProp` macro and their geometrical meaning are shown below for both variants:



`\FAProp(f_x, f_y)(f_x, f_y)(c_x, c_y){g}{a}`



`\FAProp(f_x, f_y)(t_x, t_y)(\kappa,){g}{a}`

The latter two arguments, *g* and *a*, respectively determine line and arrow style:

———— *g* = Straight

———— *a* = 0

- - - - - *g* = ScalarDash

————▶ *a* = 1

⋯⋯⋯ *g* = GhostDash

————◀ *a* = -1

~~~~~ *g* = Sine

~~~~~ *g* = Cycles

Note the slash (/) in the line-style directive which is necessary because the directive is directly handed to the PostScript interpreter.

E.3 Vertices

Vertices mark the points where propagators join. Each propagator has a counter-term order associated with it.



`\FAVert(x, y){o}` *o* = ⋯ -3 -2 -1 0 1 2 3 ⋯

E.4 Labels

Labels are usually associated with propagators, but can in principle be set anywhere. They are positioned with a pair of coordinates and an alignment, given in the usual \TeX manner, i.e. a code of up to two letters for vertical and horizontal alignment: $\{\text{t} = \text{top}, (\text{empty}) = \text{center}, \text{b} = \text{bottom}\} \otimes \{1 = \text{left}, (\text{empty}) = \text{center}, \text{r} = \text{right}\}$, e.g. `[t]` or `[rb]`. The alignment makes it possible to change the label's text, in particular its width, without having to reposition the coordinates.

$$\backslash\text{FALabel}(x,y)[align]\{text\}$$

F Incompatible Changes in Version 3.3

From Version 3.2 on, the topology shapes are accessed in a slightly different naming scheme. This eliminates some problems with large directories in some operating systems. Shapes from *FeynArts* 3.1 can be converted as follows: Replace the ShapeData directory that comes with the current *FeynArts* with your old (3.1) ShapeData directory. Then start *Mathematica* and type << Convert31to32.m. For example,

```
cd FeynArts-n.m
rm -fr ShapeData
cp -rp ../FeynArts-3.1/ShapeData .
math
<< Convert31to32.m
```

Acknowledgements

The graphics routines of *FeynArts* 3 have been developed on a Visiting Scholar grant of Wolfram Research, Inc., and I am grateful in particular to Michael Malak and Lou D'Andria for sharing their knowledge of *Mathematica* with me. Thanks also to the Ph.D. students at Karlsruhe, especially Christian Schappacher, for relentless beta testing.

References

- [De93] A. Denner, *Fortschr. Phys.* **41** (93) 307 [arXiv:0709.1075].
- [De95] A. Denner, S. Dittmaier, G. Weiglein, *Nucl. Phys.* **B440** (95) 95 [arXiv:hep-ph/9410338].
- [De92] A. Denner, H. Eck, O. Hahn, J. Küblbeck, *Nucl. Phys.* **B387** (92) 467.
- [Eck95] H. Eck, Ph.D. thesis, University of Würzburg, 1995, available from <http://www.feynarts.de>.
- [Hal06] T. Hahn, J.I. Illana, *Nucl. Phys. Proc. Suppl.* **160** (2006) 101 [hep-ph/0607049].
- [Gu86] J.F. Gunion, H.E. Haber, *Nucl. Phys.* **B272** (1986) 1.
- [HHG90] J.F. Gunion, H.E. Haber, G. Kane, S. Dawson, *The Higgs Hunter's Guide*, Frontiers in Physics Vol. 80, Addison-Wesley, 1990.
- [Ha85] H.E. Haber, G. Kane, *Phys. Rep.* **117** (1985) 75.
- [HaS02] T. Hahn, C. Schappacher, *Comp. Phys. Commun.* **143** (2002) 54 [hep-ph/0105349].
- [Ho00] W. Hollik, E. Kraus, D. Stöckinger, *Eur. Phys. J.* **C23** (2002) 735 [arXiv:hep-ph/0007134].
- [Kü90] J. Küblbeck, M. Böhm, A. Denner, *Comp. Phys. Commun.* **60** (90) 165.

Index

.Xdefaults, 35
\$DisplayFunction, 35
\$ExcludeTopologies, 16
\$FermionLines, 49
\$ModelDebug, 67
\$ModelDebugForm, 67
\$SVMixing, 21, 74
\$ShapeDataDir, 39
\$Verbose, 7
\FADiagram, 81
\FALabel, 81
\FAProp, 81
\FAVert, 81
1PI, 12, 58

ABORT, 39
accent on label, 61
Adjacencies, 11
adjacency, 9, 11
Af, 77
alignment, 37
All, 11
AllBoxCTs, 14
AllBoxes, 14
Alph, 62
amplitude, 40
 structure of, 47
AmplitudeLevel, 42, 46
AnalyticalCoupling, 53
AnalyticalPropagator, 51
animation, 35
antiparticle, 20
ANY POSITION, 39

Appearance, 62
attributes
 classes, 58
AutoEdit, 34, 35, 37
Automatic, 24
autoscaling, 35

background field, 76
bar on label, 61
Bicycle, 17
BoxCTs, 14
BoxCTsOnly, 16
Boxes, 14
BoxesOnly, 16
buttons, 39

C, 63
C2A, 77, 80
C2B, 77, 80
CA, 77, 80
CAB, 77, 80
CANCEL, 39
CB, 77, 80
CBA, 77, 80
CC, 63
Centre, 17
changes in *FeynArts* 3.3, 85
character conversion, 62
Charge, 60
charge conjugation, 20
charginos, 77
chirality projector, 21, 55
ChiralityProjector, 71
CKM, 73

- Classes, 20
- classes model, 25, 57
- closure, 54
- ColumnsXRows, 34
- combinatorial factor, 45
- compatibility, 49, 85
- complement, 32
- ComposedChar, 61
- composite label, 61
- Conjugate, 56
- ConjugateCoupling, 63
- conjugation, 65
- conservation of quantum numbers, 13, 58, 60
- conventions (*FeynArts* 1), 49
- counter-terms, 11, 63, 64
 - order of, 11
- coupling
 - antisymmetric, 54
 - classes, 63
 - closure, 54
 - conjugated, 65
 - constant, 53
 - generic, 53
 - kinematical structure, 21
 - quartic gauge-boson, 53, 63
 - structure, 45
 - tree-level, 63
 - triple gauge-boson, 54
 - vector, 53, 63
- CouplingVector, 69
- CreateCTTopologies, 11
- CreateFeynAmp, 42
- CreateTopologies, 10
- CreateVFTopologies, 12
- CT, 17
- CTOrder, 11
- CW, 73
- Cycles, 61
- dCKM1, 74
- dCW1, 74
- define, 19
- diagram
 - complement, 32
- DiagramComplement, 32
- DiagramGrouping, 29
- DiagramMap, 32
- DiagramSelect, 28
- DiracMatrix, 69, 71
- DiracSlash, 71
- DiracSpinor, 53, 56, 71
- directory contents, 7
- Discard, 31
- Display, 36
- DisplayFunction, 34
- dMf1, 74
- dMHsq1, 74
- dMWsq1, 74
- dMZsq1, 74
- drawing diagrams, 34
- dSW1, 74
- DumpGenericModel, 67
- DumpModel, 67
- dvips, 81
- dZAA1, 74
- dZAZ1, 74
- dZe1, 74
- dZfL1, 74
- dZfR1, 74
- dZG01, 74
- dZGp1, 74

- dZH1, 74
- dZW1, 74
- dZZA1, 74
- dZZZ1, 74
- Eight, 17
- EL, 73
- electroweak Standard Model, 73
- ExcludeFieldPoints, 24, 26
- ExcludeParticles, 24, 26
- ExcludeTopologies, 11, 13
- Export, 36
- External, 9, 41, 51, 61
- external
 - legs, 10
 - wave functions, 42
- F, 20
- \FADiagram, 37
- \FALabel, 37
- \FAProp, 37
- \FAVert, 37
- fermion, 20
 - amplitude, 48
 - flipping rules, 54
 - flow, 48
 - Majorana, 71
 - self-conjugate, 71
 - signs, 48
 - type, 74
- FermionChain, 41
- FermionRouting, 31
- FeynAmp, 40
- FeynAmpCases, 31
- FeynAmpDenominator, 41, 45
- FeynAmpList, 40
- FeynArts* 1, 49
- feynarts.sty*, 36, 81
- FeynArtsGraphics, 36
- feynartspicture, 37, 81
- Feynman gauge, 42
- Feynman rules, 68
- FeynmanRules, 69
- Field, 28
- field, 20
 - background, 76
 - derived, 26
 - exclusion, 26
 - extended structure, 52, 53
 - insertion, 20
 - intermediate, 52
 - level, 20
 - lower level, 26
 - mixing, 20
- FieldMatchQ, 30
- FieldMemberQ, 30
- FieldNumbers, 28, 34
- FieldPointMatchQ, 30
- FieldPointMemberQ, 30
- FieldPoints, 30
- filter, 13
 - function, 16
- final-touch rules, 56, 66
- flipping rules, 54
- Format, 50
- FourMomentum, 41, 71
- FourVector, 69, 71
- frame rate, 35
- frameTime, 35
- G, 41, 53
- gauge, 42
- GaugeRules, 42

- GaugeXi, 41, 42, 51, 73
- Generation, 57, 58, 62
- Generic, 20
- generic coupling, 41, 45
- generic model, 25, 50
- GenericModel, 24, 26
- ghost field, 20
- GhostDash, 61
- Global', 19
- gluinos, 77
- gluon, 75
- gluon line, 61
- graphics primitives, 81
- GraphID, 40
- Greek, 62
- GRID POSITION, 39
- GS, 75
- h.c., 65
- HexagonCTsOnly, 16
- HexagonsOnly, 16
- ImageSize, 36
- Incoming, 9
- Index, 52, 57
- index
 - colour, 61
 - Dirac, 49
 - generation, 21, 57, 58, 62
 - Greek, 62
 - kinematic, 50
 - Latin, 62
 - Lorentz, 50, 62
 - range of, 57
 - summed over, 41
- IndexDelta, 65
- IndexRange, 57
- IndexSum, 41
- Indices, 59
- initialization file, 7
- InitializeModel, 25
- InsertFields, 22
- insertion
 - auxiliary functions, 29
 - discarding an, 31
 - grouping, 29
 - modifying an, 32
 - selecting, 28
 - structure of, 33
- InsertionLevel, 24
- InsertOnly, 59
- Installation, 6
- Integral, 40
- integration momenta, 40
- interaction eigenstates, 60
- Internal, 9, 14, 51, 61
- irreducible, 12, 14
- J/Link, 39
- Java, 39
- kinematic object, 41, 45
- kinematic vector, 53, 63, 65
- KinematicIndices, 50
- KinematicVector, 65
- label, 61
 - \LaTeX , 37, 61
 - composite, 61
 - indices, 62
 - of mixing field, 63
- Lagrangian, 69
- LastSelections, 24, 27
- \LaTeX commands, 81

- leptons, 74
- level, 20, 47
 - classes, 20
 - generic, 20
 - particles, 20
 - picking a, 40, 46
 - specification of, 7, 22
- linear combination, 60
- LoadModel, 66
- Loop, 9
- LoopFields, 29
- LoopNumber, 42
- loops
 - irreducible conglomerate of, 9, 19
 - number of, 10
 - quark, 61
- Lorentz, 51, 62
- Lorentz formalism, 21
- Lorentz.gen, 50, 69, 71
- Lorentzbgf.gen, 76
- M\$ClassesDescription, 57, 58
- M\$CouplingMatrices, 57, 63, 70
- M\$FlippingRules, 50, 55
- M\$GenericCouplings, 50, 53
- M\$GenericPropagators, 50, 51
- M\$LastGenericRules, 50, 56, 71
- M\$LastModelRules, 57, 66
- M\$TruncationRules, 42, 50, 56
- MA0, 77, 80
- MajoranaSpinor, 53, 56, 71
- Mass, 51, 59
- mass eigenstates, 60
- MatrixTrace, 41
- MatrixTraceFactor, 59
- MB, 73
- MC, 73
- MCha, 77
- MD, 73
- MDQ, 73
- ME, 73
- MetricTensor, 52, 71
- MG0, 42, 73, 77, 80
- MGL, 77
- MGp, 42, 73, 77, 80
- MH, 73
- Mh0, 77, 80
- MHH, 77, 80
- MHp, 77, 80
- mixing field, 20, 59
 - reversed, 20
- MixingPartners, 59
- Mixture, 59
- mixture, 60
- ML, 73
- MLE, 73
- MM, 73
- MNeu, 77
- Model, 24
- model, 25
 - classes, 25, 57
 - definition, 50
 - generation of, 68
 - generic, 25, 50
 - initialization, 25
 - MSSM, 77
 - template file, 69
 - two-Higgs-doublet, 80
- ModelDebug, 67
- ModelEdit, 26
- ModelMaker*, 68
- momenta, 49

- MomentumConservation, 42
- motifs, 35
- MQU, 60, 73
- MS, 73
- MSf, 77
- MSSM, 77
- MSSM.mod, 77
- MSSMQCD.mod, 77
- MT, 73
- MU, 73
- MUE, 77
- multiplet, 57
- MW, 42, 73
- MZ, 42, 73
- neutralinos, 77
- neutrinos, 74
- NoElectronHCoupling, 27, 75, 79, 80
- NoGeneration n , 75, 79
- NoLightFHCoupling, 75, 79, 80
- NonCommutative, 52
- noncommuting object, 41, 52
- NoQuarkMixing, 75
- NoSUSYParticles, 79
- NoUnfold, 57
- Numbering, 34
- numbering
 - for use with Discard, 32, 34
- OK, 39
- Outgoing, 9
- Paint, 34
- PaintLevel, 34
- Particles, 20
- patterns, 51, 64
- PentagonCTsOnly, 16
- PentagonsOnly, 16
- permutable, 18
- photon line, 61
- PickLevel, 40, 46
- placeholders, 12
- Poincaré group, 50
- polarization vector, 45
- PolarizationVector, 52, 56, 71
- PostScript, 81
- PreFactor, 42
- prefactor, 45
- Propagator, 9
- propagator, 9
 - generic, 51
 - type of, 9
 - vector-boson, 52
- PropagatorArrow, 59
- PropagatorDenominator, 41, 51
- PropagatorLabel, 59
- PropagatorType, 59
- QCD, 75
- QED.gen, 50
- QEDOnly, 75
- quantum numbers, 13, 58, 60
- QuantumNumbers, 59
- quarks, 74
- Reinitialize, 26
- RelativeCF, 45
- renormalization constants, 74
- replacement rules, 45–47
- Restrictions, 24, 27, 57, 75
- REVERT, 39
- roadmap, 8
- RParity, 60

- S, 20
- S2A, 77, 80
- S2B, 77, 80
- SA, 77, 80
- SAB, 77, 80
- SB, 77, 80
- SBA, 77, 80
- scalar field, 20
- scalar–vector mixing, 20
- ScalarDash, 61
- screen messages, 7
- SelfConjugate, 59, 71
- SelfEnergies, 14
- SelfEnergiesOnly, 16
- SelfEnergyCTs, 14
- SelfEnergyCTsOnly, 16
- Setup.m, 7
- sfermions, 77
- Shape, 37
- shapes, 35
 - saving, 39
- SheetHeader, 34
- Show, 36
- Sine, 61
- SM.mod, 73
- SMbgf.mod, 76
- SMc.mod, 75
- SMQCD.mod, 75
- specifying momenta, 49
- Standard Model, 73
 - masses, 73
- starting topology, 11, 17
 - definition, 18
 - name, 19
- StartingTopologies, 11, 17
- StartTop, 17
- Straight, 61
- subscript, 61
- SumOver, 41
- SUNF, 75
- SUNT, 75
- superscript, 61
- supersymmetry, 77
- SUSY, 77
- SV, 20, 51
- SW, 73
- SymmetryFactor, 17, 19
- T, 20
- TadpoleCTs, 14
- TadpoleCTsOnly, 16
- Tadpoles, 14
- TadpolesOnly, 16
- TB, 77, 80
- template model file, 69
- tensor field, 20
- THDM, 80
- THDM.mod, 80
- THDMParticles, 79
- TheLabel, 62
- TheMass, 60
- Theta, 17
- Three, 17
- ThreeRed, 17
- ToFA1Conventions, 49
- Topology, 10
- topology, 9
 - comparison, 18
 - connected, 9
 - counter-term, 11
 - editor, 38
 - exclusion of, 13

- filter, 13
 - defining own, 16
 - function, 16
- inserted, 33
- irreducible, 14
- starting, 11, 17
 - definition, 18
 - name, 19
- with vertex function, 12
- Topology.m, 19
- TopologyEditor.java, 39
- TopologyList, 10
- ToTree, 17
- TriangleCTs, 14
- TriangleCTsOnly, 16
- Triangles, 14
- TrianglesOnly, 16
- Truncated, 42
- U, 20
- UCA α , 62
- UCGreek, 62
- UCh α , 77
- USf, 77
- V, 20
- V4onExt, 16
- VCh α , 77
- vector field, 20
- Vertex, 9
- vertex, 9
 - adjacency, 9, 11
 - identifiers, 19
 - permutable, 18
- vertex functions, 12
- VertexFunction, 12, 41
- Vertices, 30
- ViolatesQ, 57, 58, 60
- VS, 21
- wave-function corrections, 14
- WFCorrectionCTFields, 29
- WFCorrectionCTs, 14
- WFCorrectionFields, 29
- WFCorrections, 14
- WriteModelFile, 69
- ZNeu, 77